Министерство образования и науки Украины
Харьковский национальный университет имени В. Н. Каразина

Т.В. Клочко, Н.Д. Парфёнова

РЕШЕНИЕ ЗАДАЧ КОМПЛЕКСНОГО АНАЛИЗА СРЕДСТВАМИ Maple

Харьков – 2009

УДК 51-3:517.53(075.8)
ББК 22.161.5я73 + 32.973-018.2я73
К 50
Утверждено на заседании научно-методического совета Харьковского национального университета имени В.Н. Каразина (протокол № от .09.2009).

Рецензенты:

доктор физико-математических наук, профессор
Фаворов Сергей Юрьевич,
Харьковский национальный университет имени В.Н. Каразина,
заведующий кафедрой теории функций и
функционального анализа

кандитат физико-математических наук, доцент
Зиненко Сергей Николаевич,
Харьковский национальный университет имени В.Н. Каразина,
кафедра высшей математики физического факультета

К 50 **Клочко Т. В., Парфенова Н. Д.** Решение задач комплексного анализа средствами Maple: учебно-методическое пособие. – Х.: ХНУ имени В. Н. Каразина, 2009. – 68 с.

В учебно-методическом пособии рассматриваются типовые задачи из курса комплексного анализа, который изучается в 4 семестре студентами 2 курса физического и радио-физического факультетов Харьковского национального университета имени В.Н. Каразина, и их решения с помощью системы компьютерной математики - Maple. Материал разбит на 7 лабораторных занятий по данному курсу. Особое внимание уделено технике практических вычислений и визуализации их результатов. Каждая тема имеет задачи для индивидуальной работы каждого студента.

УДК 51-3:517.53(075.8)
ББК 22.161.5я73 + 32.973-018.2я73

СОДЕРЖАНИЕ

Введение

Внедрение современных технологий в процесс обучения способствует повышению уровня информационной культуры студентов. Использование компьютерных технологий при изучении нового материала в дальнейшем приводит к лучшему его усвоению, приближает процесс обучения к научному поиску.

Рассмотрим пример изучения пакета Maple в курсе "Математического моделирования" на задачах из параллельного курса "Комплексный анализ". Такой подход позволяет продемонстрировать возможности изучаемого математического пакета и при этом закрепить методы решения задач классического математического курса; наглядно дает увидеть межпредметные связи далеко не очевидные студентам.

"Комплексный анализ" является важным разделом курса высшей математики любого естественнонаучного направления, он необходим для усвоения более специальных предметов – математической и теоретической физики, гидродинамики, аэродинамики. Чтобы успешно освоить этот курс студентам необходимо решить большое количество задач. Однако в стандартных курсах обучению методам решения задач обычно уделяется недостаточно времени и поэтому для студентов со средней математической подготовкой решение несложных практических задач может вызывать большие затруднения.

Не смотря на то, что пакеты типа Maple работают по принципу "черного ящика": на "входе" — исходные данные, на "выходе" — результат (при этом ход решения задачи пользователю может быть и неизвестен); возможно приобрести умение его использовать и повторить аналитические методы решения задач. В этом и состоит искусство преподавателя подбирающего конкретные задачи конкретным студентам. Правильный подбор задач позволяет повторить изученный метод и проанализировать полученное решение.

В данном учебно-методическом пособии, применяя только самые необходимые средства Maple, решаются типовые задачи комплексного анализа. Описания и полные списки параметров встроенных функций, как правило, не приводятся. С ними, как и с теми средствами Maple, которые не вошли в данное руководство, желающие могут ознакомиться как в справочной системе Maple, так и в более подробных руководствах, посвященных Maple (см. [1]-[4]). Задачи разбиты на 7 тем и решаются, в такой последовательности, в какой они изучаются в курсе комплексного анализа. Решения задач в Maple сопровождаются комментариями. Данное учебно-методическое пособие задуманно как дополнение к учебным курсу комплексного анализа. Решения задач в Maple сопровождаются комментариями, приводятся полные тексты программ Maple для их решения.

1 Комплексные числа

1.1 Арифметические операции с комплексными числами

Комплексные числа $x + iy$ вводятся в командную строку в виде $x + y * I$. Например,

```
> z1:=-4-3*I; z2:=-1+2*I;
```

$$z1 := -4 - 3\,I$$

$$z2 := -1 + 2\,I$$

Действительная и мнимая части комплексного числа (функции комплексной переменной) находятся функциями `Re(z)` и `Im(z)`, соответственно:

```
> `Re z1`:=Re(z1); `Im z1`:=Im(z1);
```

$$Re\ z1 := -4$$

$$Im\ z1 := -3$$

Комплексно-сопряженное число выдает функция `conjugate(z)`:

```
> conjugate(2+3*I);
```

$$2 - 3I$$

Алгебраические действия с комплексными числами показаны в следующем примере:

Пример 1. *Выполнить арифметические действия над комплекными числами в алгебраической форме:* $z_1 = -4 - 3i; z_2 = -1 + 2i$.

```
> `z1+z2`:=z1+z2; `z1-z2`:=z1-z2;
> `z1 * z2`:=z1*z2; `z1 / z2`:=z1/z2;
```

$$z1 + z2 := -5 - I$$

$$z1 - z2 := -3 - 5I$$

$$z1 * z2 := 10 - 5I$$

$$z1/z2 := \frac{-2}{5} + \frac{11}{5}I$$

1.2 Различные формы комплексных чисел

Модуль и главное значение аргумента[1] комплексного числа вычисляются встроенными функциями `abs(z)` и `argument(z)`, соответственно. Например,

```
> z:=1+sqrt(3)*I;
```

$$z := 1 + I\sqrt{3}$$

[1]Обратите внимание на то, что Maple под главным значением аргумента комплексного числа понимает значение из промежутка $(-\pi; \pi]$.

```
> abs(z); argument(z);
```

$$2$$

$$\frac{1}{3}\pi$$

Они одновременно выводятся встроенной функцией polar(z):

```
> polar(1 + I*sqrt(3));
```

$$\mathrm{polar}\left(2, \frac{1}{3}\pi\right)$$

Пример 2. *Найти вещественную и мнимые части, модуль и главное значение аргумента числа $z = -1 + \sqrt{3}i$. Изобразить.*

Задаем значение z и выводим число z, его модуль $|z|$, главное значение аргумента $\arg z$, вещественную и мнимую части $\mathrm{Re}\ z$, $\mathrm{Im}\ z$. Для улучшения внешнего вида выводимой информации здесь и в дальнейшем будут часто использованы различные виды кавычек (подробнее см. [4] с. 20-24).

```
> z:=-1+sqrt(3)*I;
> `|z|`=abs(z); `arg z` =argument(z);
> x:=Re(z): y:=Im(z): `Re z` =Re(z); `Im z` =Im(z);
```

$$z := -1 + \sqrt{3}I$$

$$|z| = 2$$

$$arg\ z = \frac{2\pi}{3}$$

$$Re\ z = -1$$

$$Im\ z = \sqrt{3}$$

Системы аналитических вычислений, и Maple здесь не исключение, привлекают многих своих пользователей удобными и хорошо реализованными возможностями отображения графической информации. Ориентированные на выполнение математических преобразований в аналитической форме они предоставляют инструменты для работы с разнообразными геометрическими объектами.

Используя графические возможности Maple (при необходимости желающие могут с ними ознакомится по книгам [1]– [4]), получаем нужное изображение.

```
> with(plots): with(plottools):
> p1:=pointplot([x,y],color=black,symbol=CIRCLE, symbolsize=14):
> p2:=arrow([0,0],[x,y],0.05,0.15,0.1,color=black):
> p3:=plot(y,x..0,linestyle=2,color=grey):
> p4:=plot([x,0],[x,y],linestyle=2,color=grey):
```

```
> p5:=polarplot(abs(z),color=grey):
> p6:=arc([0,0], 0.5, 0..`arg z`):
> p7:=textplot([-x*0.3,y*0.3,"arg z"],align=RIGHT):
> p8:=textplot([x-0.05,y+0.05,"z"],align={ABOVE,LEFT}):
> p9:=plot(0,-abs(z)..abs(z),linestyle=2,color=grey):
> p10:=textplot([x-0.05,0.05,"Re z"],align=ABOVE):
> p11:=textplot([0.05,y,"Im z"],align=RIGHT):
> display(seq(p||i,i=1..11),scaling=constrained);
```



Пример 3. *Найти тригонометрическую и показательную формы комплексного числа заданного в алгебраической форме:* $z = -1 + i$.

Зададим комплексное число z в алгебраической форме и выведем его.

```
> z:=-1+I;
```

$$z := -1 + I$$

Находим модуль и главное значение аргумента числа z.

```
> `|z|`:=abs(z); `arg z`:=argument(z);
```

$$|z| := \sqrt{2}$$

$$\arg z := \frac{3\pi}{4}$$

Выводим комплексное число z в показательной форме:

```
> z:= `|z|`*'exp'(I*`arg z`);
```

$$z := \sqrt{2}e^{\left(\frac{3}{4}I\pi\right)}$$

Выводим комплексное число z в тригонометрической форме:

```
> z:=`|z|`*('cos'(`arg z`)+I*'sin'(`arg z`));
```

$$z := \sqrt{2}\left(\cos\left(\frac{3\pi}{4}\right) + \sin\left(\frac{3\pi}{4}\right) I\right)$$

1.3 Натуральные степени и корни

Как известно, возводить комплексные числа в степень, а так же извлекать корни, удобнее в тригонометрической или в показательной форме.

Пример 4. *Найти n-ую степень комплексного числа z. Изобразить z и z^n.*
Задаем значения z и n :

```
> z:=sqrt(3)/2+I; n:=3;
```

$$z := \frac{\sqrt{3}}{2} + I$$

$$n := 3$$

Находим модуль и главное значение аргумента числа z, вещественную и мнимую части $\operatorname{Re} z$, $\operatorname{Im} z$, необходимые для дальнейшего построения.

```
> x:=Re(z); y:=Im(z); `|z|`=abs(z); `arg z`:=argument(z);
```

$$x := \frac{\sqrt{3}}{2}$$

$$y := 1$$

$$|z| = \frac{\sqrt{7}}{2}$$

$$\arg z := \arctan\left(\frac{2\sqrt{3}}{3}\right)$$

Вычисляем n-ую степень z.

```
> w:= z^n ;
```

$$w := \left(\frac{\sqrt{3}}{2} + I\right)^3$$

Значения функций комплексной переменной находятся функцией `evalc`.

```
> w:=evalc(z^n);
```

$$w := -\frac{9\sqrt{3}}{8} + \frac{5}{4}I$$

Находим модуль и главное значение аргумента числа $w = z^n$, вещественную и мнимую части $\operatorname{Re} w$, $\operatorname{Im} w$. Выводим их.

```
> xw:=Re(w); yw:=Im(w); `|w|`:=abs(w); `arg w`:=argument(w);
```

$$xw := -\frac{9\sqrt{3}}{8}$$

$$yw := \frac{5}{4}$$

$$|w| := \frac{7\sqrt{7}}{8}$$

$$\arg w := -\arctan\left(\frac{10\sqrt{3}}{27}\right) + \pi$$

Используя стандартные возможности Maple, получаем изображение.

```
> with(plots): with(plottools):
> p1:=arrow([0,0],[x,y],0.05,0.15,0.1,color=black):
> p2:=arc([0,0],abs(z)/2, 0..argument z):
> p3:=arc([0,0],abs(z)/2+0.05, 0..argument z):
> p4:=textplot([x*0.8,y*0.5,"arg z"],align=RIGHT):
> p5:=textplot([x-0.05,y+0.05,"z"],align={ABOVE,LEFT}):
> p6:=arrow([0,0],[xw,yw],0.05,0.15,0.1,color=black):
> p7:=arc([0,0], abs(w)/10, 0..argument z):
> p8:=textplot([xw*0.035,yw*0.4,"arg z^n"],align=LEFT):
> p9:=textplot([xw+0.1,yw+0.1,"z^n"],align={ABOVE,RIGHT}):
> p10:=polarplot(abs(z),color=grey):
> p11:=polarplot(abs(w),color=grey):
> display(seq(p||i,i=1..11));
```



Пример 5. *Найти корни* $\sqrt[n]{z}$ *и изобразить их:* $n = 4,\ z = 1 + \sqrt{3}i$.

Прежде всего вспомним формулу для вычисления корней $\sqrt[n]{z}$:

```
> w=abs(z)^(1/n)*(cos((arg(z)+2*pi*k)/n)+I*sin((arg(z)+2*pi*k)/n));
> `k=0,...,n-1`;
```

$$w = |z|^{\left(\frac{1}{n}\right)} \left(\cos\left(\frac{\arg(z) + 2\pi k}{n}\right) + \sin\left(\frac{\arg(z) + 2\pi k}{n}\right) I\right)$$

$$k = 0, \ldots, n - 1$$

Задаем значения z и n :

```
> n:=4; z:=1+sqrt(3)*I;
```

$$n := 4$$

$$z := 1 + \sqrt{3}I$$

Функция `evalc` возвращает только главное значение $\sqrt[n]{z}$.

```
> evalc(z^(1/n));
```

$$2^{\left(\frac{1}{4}\right)} \cos\left(\frac{\pi}{12}\right) + 2^{\left(\frac{1}{4}\right)} \sin\left(\frac{\pi}{12} I\right)$$

Все значения $\sqrt[n]{z}$ находятся функцией `fsolve`, как корни уравнения $w^n = z$. В частности:

```
> s:=fsolve(w^n=z,w,complex);
```

$$s := -1.148685865 - .3077894500I,\ -.3077894500 + 1.148685865I,$$
$$.3077894500 - 1.148685865I,\ 1.148685865 + .3077894500I$$

Результат получился неприемлимый в некоторых ситуациях, хотелось бы получить точные значения всех корней. Поэтому, зная формулу для вычисления корней (она была приведена в начале решения этого примера), определим функцию $w = \sqrt[n]{z}$ и вычислим корни с ее помощью.

```
> w:=(r,phi,k,n)->
  (r)^(1/n)*('cos'((phi+2*Pi*k)/n)+I*'sin'((phi+2*Pi*k)/n)):
> r:=abs(z):phi:=argument(z):
> w0:=w(r,phi,0,n);w1:=w(r,phi,1,n);w2:=w(r,phi,2,n);w3:=w(r,phi,3,n);
```

$$w0 := 2^{\left(\frac{1}{4}\right)}\left(\cos\left(\frac{\pi}{12}\right) + \sin\left(\frac{\pi}{12}\right)I\right)$$

$$w1 := 2^{\left(\frac{1}{4}\right)}\left(\cos\left(\frac{7\pi}{12}\right) + \sin\left(\frac{7\pi}{12}\right)I\right)$$

$$w2 := 2^{\left(\frac{1}{4}\right)}\left(\cos\left(\frac{13\pi}{12}\right) + \sin\left(\frac{13\pi}{12}\right)I\right)$$

$$w3 := 2^{\left(\frac{1}{4}\right)}\left(\cos\left(\frac{19\pi}{12}\right) + \sin\left(\frac{19\pi}{12}\right)I\right)$$

Строим изображение корней n-ой степени комплексного числа z используя функцию `polygon` для получения правильного n-угольника.

```
> A:=[Re(s[4]),Im(s[4])]: B:=[Re(s[2]),Im(s[2])]:
> C:=[Re(s[3]),Im(s[3])]: E:=[Re(s[1]),Im(s[1])]:
> p0:=polarplot(abs(s[1]),color=grey):
> p1:=polygon([A,B,E,C,A], thickness=1):
> for i from 1 to n by 1 do
> pict[i]:=arrow([0,0],[Re(s[i]),Im(s[i])],0.02,0.05,0.1,color=black):
> end do:
> p2:=textplot([Re(s[4]),Im(s[4])+0.1,"w0"],align=ABOVE):
> p3:=textplot([Re(s[2]),Im(s[2])+0.05,"w1"],align=ABOVE):
> p4:=textplot([Re(s[1])-0.2,Im(s[1]-0.2),"w2"],align=RIGHT):
> p5:=textplot([Re(s[3])-0.05,Im(s[3]),"w3"],align=LEFT):
> display(seq(pict[i],i=1..n),seq(p||i,i=0..5));
```



1.4 Задачи для самостоятельного решения

1. Выполнить арифметические действия над комплекными числами в алгебраической форме:

1)	$z_1 = -1 - i;$	$z_2 = -1 + 2i.$	2)	$z_1 = 1 - i;$	$z_2 = -1 + 2i.$
3)	$z_1 = -\sqrt{3} + i;$	$z_2 = -1 + 7i.$	4)	$z_1 = \sqrt{3} + i;$	$z_2 = -7 + 2i.$
5)	$z_1 = -2 - 2i;$	$z_2 = -1 + i.$	6)	$z_1 = 2 - 2i;$	$z_2 = -1 + 4i.$
7)	$z_1 = -3 - 3i;$	$z_2 = -1 + 4i.$	8)	$z_1 = 3 - 3i;$	$z_2 = -4 + i.$
9)	$z_1 = -1 + \sqrt{3}i;$	$z_2 = -2 + 2i.$	10)	$z_1 = 1 + \sqrt{3}i;$	$z_2 = 1 + 2i.$
11)	$z_1 = -1 - \sqrt{3}i;$	$z_2 = -3 + 7i.$	12)	$z_1 = 1 - \sqrt{3}i;$	$z_2 = 7 + 2i.$
13)	$z_1 = -\sqrt{3} - i;$	$z_2 = -4 + i.$	14)	$z_1 = \sqrt{3} - i;$	$z_2 = 1 + 4i.$
15)	$z_1 = -4 - 4i;$	$z_2 = -7 + i.$	16)	$z_1 = 4 - 4i;$	$z_2 = -6 + 4i.$

2. Найти вещественную и мнимые части, модуль и главное значение аргумента числа z_1. Изобразить.

3. Найти тригонометрическую и показательную формы числа z_1.

4. Найти n-ую степень комплексного числа z_1. Изобразить z_1 и z_1^n.

1)	$n = 3;$	2)	$n = 4;$	3)	$n = 5;$	4)	$n = 6;$
5)	$n = 3;$	6)	$n = 4;$	7)	$n = 5;$	8)	$n = 6;$
9)	$n = 3;$	10)	$n = 4;$	11)	$n = 5;$	12)	$n = 6;$
13)	$n = 3;$	14)	$n = 4;$	15)	$n = 5;$	16)	$n = 6;$

5. Найти корни $\sqrt[n]{z_1}$ и изобразить их.

2 Геометрическая интерпретация комплексных чисел

Замечание. В комплексном анализе принято штриховать дополнение искомой области, а не ее саму. Поэтому заливку серым будем делать той части плоскости, которая не удовлетворяет условиям.

2.1 Примеры

Задачи вида "Выяснить геометрический смысл соотношений" очень трудно алгоритмизиравать и даже классифицировать. В них необходимы хотя бы элементарные геометрические знания. Большая часть этих задач будут решаться с использованием либо определения кривых, либо функции Maple `conic`, которая создана для работы с произвольными кривыми второго порядка, то есть линий, определяемых уравнением вида

$$ax^2 + bxy + cy^2 + dx + ey + f = 0.$$

Подробнее о функции `conic` можно прочесть, например, в [4], а классификация кривых второго порядка была рассмотрена в курсе аналитической геометрии (например, см. [5]).

Если граничная линия принадлежит области (нестрогое неравенство), будем изображать ее сплошной линией (linestyle=1), если не принадлежит (строгое неравенство) – пунктирной (linestyle=3).

Пример 6. *Выяснить геометрический смысл соотношения и изобразить:*

Окружность: $|\mathbf{z} - \mathbf{z_0}| = \mathbf{R}$

Зададим условия задачи:

```
> z0:=1-0.5*I: R:=3: abs(z-z0)=R; z:=x+y*I:
```

$$|z - 1. + 0.5I| = 3$$

Найдем образ данного соотношения, используя функцию implicitplot, предназначенную для изображения неявно заданных функций.

```
> with(plottools): with(plots):
> c:=implicitplot(abs(z-z0)=R, x=-5..5,y=-5..5,color=black,
  thickness=3):
> p0:=pointplot([Re(z0),Im(z0)],color=black,symbol=CIRCLE,
  symbolsize=6):
> p1:=pointplot([Re(z0),Im(z0)],color=black,symbol=CIRCLE,
  symbolsize=14):
> p2:=textplot([Re(z0),Im(z0)-0.25,"z0"],align=BELOW):
> display([c,p0,p1,p2]);
```



Видим, что получилась окружность, попытаемся найти ее уравнение.

```
> evalc(abs(z-z0)=R); lhs(%)^2=R^2;
```

$$\sqrt{(x-1.)^2+(y+0.5)^2}=3$$

$$(x-1.)^2+(y+0.5)^2=9$$

Таким образом, $|z-z_0|=R$ – это окружность с центром в точке z_0 радиуса R. Ее каноническое уравнение:

$$(x-\mathrm{Re}(z_0))^2+(y-\mathrm{Im}(z_0))^2=R^2.$$

Кольцо: $\mathbf{r \leq |z-z_0| < R}$.

Учитывая предыдущий пример, понятно, что границей заданной области будут окружности. Таким образом, нас интересуют все точки плоскости, которые находятся на расстоянии большем или равном, чем r и меньшем чем R. Причем меньшая окружность принадлежит области, ее изобразим сплошной линией, а большая не принадлежит, значит, ее изобразим пунктирной.

```
> with(plots):
> r<=`|z-z0|<R`; z0:=0; r:=1; R:=2;
```

$$r <= |z-z0| < R$$
$$z0 := 0$$
$$r := 1$$
$$R := 2$$

```
> p1:=implicitplot(subs(z=x+I*y,abs(z-z0)<R),x=-3..3,y=-3..3,
  filled=true,coloring=[white,grey],linestyle=4):
> p2:=plot([sqrt(1-x^2),-sqrt(1-x^2)],x=-2..2,filled=true,
  color=[grey,grey]):
> p3:=plot([sqrt(1-x^2),-sqrt(1-x^2)],x=-2..2,y=-2..2,
  color=black,linestyle=1):
> display(p2,p3,p1);
```



Эллипс: $|\mathbf{z}-\mathbf{z_1}|+|\mathbf{z}-\mathbf{z_2}|=\mathbf{2a}$

Зададим условия задачи:

```
> z:=x+y*I: a:=4; z1:=-3+2*I; z2:=3+I;
```

$$a := 4$$

$$z1 := -3 + 2I$$

$$z2 := 3 + I$$

Найдем образ данного соотношения.

```
> with(plots): with(plottools):
> p1:=implicitplot(evalc(abs(z - z1) + abs(z - z2) = 2*a),
> x=-2*a..2*a,y=-2*a..2*a, color=black, thickness=4):
> p2:=pointplot([Re(z1),Im(z1)],color=black,symbol=CIRCLE,
  symbolsize=14):
> p3:=pointplot([Re(z2),Im(z2)],color=black,symbol=CIRCLE,
  symbolsize=14):
```

Для большей иллюстративности изобразим векторы, с началами в точках z_1 и z_2, идущие к произвольной точке z нашей фигуры.

```
> evalc(abs(z - z1) + abs(z - z2) = 2*a):
> simplify(subs(x=-2,%)):
> s:=solve(%): zx:=-2:zy:=s[1]:
> p4:=pointplot([zx,zy],color=black,symbol=CIRCLE,symbolsize=14):
> p5:=arrow([Re(z1),Im(z1)],[zx,zy],0.02,0.05,0.1,color=black):
> p6:=arrow([Re(z2),Im(z2)],[zx,zy],0.02,0.05,0.1,color=black):
> p7:=textplot([Re(z1),Im(z1)+0.3,"z1"],align=LEFT):
> p8:=textplot([Re(z2),Im(z2)+0.3,"z2"],align=RIGHT):
> p9:=textplot([zx,zy+0.4,"z"],align=LEFT):
> display(seq(p||i,i=1..9), scaling=constrained);
```



Видим, что получился эллипс, чего и следовало ожидать.Так как, множество точек z, удовлетворяющих уравнению $|z - z_1| + |z - z_2| = 2a$, где $a > \frac{1}{2}\,|z_1 - z_2|$, есть эллипс с фокусами в точках z_1, z_2 и с большой полуосью, равной a, так как $|z - z_1| + |z - z_2|$ — сумма расстояний от точки z до точек z_1 и z_2.

Замечание. Эллипсом называется геометрическое место точек, для которых сумма расстояний от двух фиксированных точек плоскости, называемых фокусами, есть постоянная величина (требуется, чтобы эта постоянная была больше расстояния между фокусами).

Гипербола: $\mathbf{||z - z_1| - |z - z_2|| = 2a}$

Зададим условия задачи:

```
> z:=x+y*I: a:=1; z1:=-3/2; z2:=3/2;
```

$$a := 1$$

$$z1 := \frac{-3}{2}$$

$$z2 := \frac{3}{2}$$

Найдем образ данного соотношения.

```
> with(plots): with(plottools):
> p1:=implicitplot(evalc(abs(abs(z -z1)-abs(z-z2))=2*a),
  x=-3*a..3*a, y=-3*a..3*a, color=black, thickness=2):
> p2:=pointplot([Re(z1),Im(z1)],color=black,symbol=CIRCLE,
  symbolsize=14):
> p3:=pointplot([Re(z2),Im(z2)],color=black,symbol=CIRCLE,
  symbolsize=14):
```

Аналогично, изобразим векторы, с началами в точках z_1 и z_2, идущие к произвольной точке z нашей фигуры.

```
> evalc(abs(abs(z - z1) - abs(z - z2)) = 2*a):
> simplify(subs(x=-2,%)): s:=solve(%): zx:=-2:zy:=evalf(s[1]):
> p4:=pointplot([zx,zy],color=black,symbol=CIRCLE,symbolsize=14):
> p5:=arrow([Re(z1),Im(z1)],[zx,zy],0.03,0.05,0.1,color=black):
> p6:=arrow([Re(z2),Im(z2)],[zx,zy],0.03,0.05,0.1,color=black):
> p7:=textplot([Re(z1)-0.1,Im(z1)+0.2,"z1"],align=LEFT):
> p8:=textplot([Re(z2),Im(z2)+0.2,"z2"],align=RIGHT):
> p9:=textplot([zx+0.1,zy+0.4,"z"],align=RIGHT):
> display(seq(p||i,i=1..9),scaling=constrained);
```



Видим, что получилась гипербола. Так как, множество точек z, удовлетворяющих уравнению $||z-z_1|-|z-z_2|| = 2a$, где $a < \frac{1}{2}|z_1-z_2|$, является уравнением гиперболы с фокусами в точках z_1, z_2, и с действительной полуосью, равной a.

Замечание. Гиперболой называется геометрическое место точек, для которых модуль разности расстояний от двух фиксированных точек плоскости, называемых фокусами, есть постоянная величина (разность должна быть меньше расстояния между фокусами и отлична от нуля).

"Внутренность гиперболы": $\mathbf{Im(z^2+z+1)<1}$

```
> restart: with(plots): with(geometry):
> eqn:=Im(z^2+z+1)-1:eqn<0;
```

$$Im(z^2+z+1) < 1$$

```
> z:=x+y*I: eqn:=evalc(eqn): eqn<0;
```

$$2xy + y < 1$$

Получилось уравнение второго порядка с двумя неизвестными, поэтому для получения полной информации воспользуемся функцией conic.

```
> conic(c1,eqn=0,[x,y]): detail(c1);
```

$name \quad of\ the\ object : c1$

$form\ of\ the\ object : hyperbola2d$

$center : [-1/2, 0]$

$foci : [[-3/2, -1], [1/2, 1]]$

$vertices :$

$[[1/2 * (-1/4 * 2^{(1/2)} - 1) * 2^{(1/2)} - 1/4,\ 1/2 * (-1/4 * 2^{(1/2)} - 1) * (1/2) + 1/4]]$

$[[1/2 * (-1/4 * 2^{(1/2)} + 1) * 2^{(1/2)} - 1/4, 1/2 * (-1/4 * 2^{(1/2)} + 1) * (1/2) + 1/4]]$

$the\ asymptotes : [y * 2^{(1/2)} = 0, -x * 2^{(1/2)} - 1/2 * 2^{(1/2)} = 0]$

$equation\ of\ the\ hyperbola : 2 * x * y + y - 1 = 0$

Таким образом, границей заданной области является гипербола, все характеристики которой были получины выше.

Теперь изобразим заданную область, обозначив на рисунке через z0 – центр гиперболы, F1 и F2 – ее фокусы, нарисуем так же и ее асимптоты.

```
> z0:=coordinates(center_c1):F1:=coordinates(foci_1_c1):
> F2:=coordinates(foci_2_c1):
> p1:=implicitplot(lhs(Equation(c1))<0,x=-3..3,y=-3..3,
  filled=true,coloring=[white,grey],linestyle=4,thickness=2):
> p2:=pointplot(z0):
> p3:=textplot([z0[1],z0[2]+0.2,"z0"],align=ABOVE):
> p4:=implicitplot(Equation(asymptote_1_c1),x=-3..3,y=-3..3,
  color=black,linestyle=1,thickness=1):
> p5:=implicitplot(Equation(asymptote_2_c1),x=-3..3,y=-3..3,
  color=black,linestyle=1,thickness=1):
> p6:=pointplot(F1): p8:=pointplot(F2):
> p7:=textplot([F1[1]-0.2,F1[2]-0.2,"F[1]"],align=BELOW):
> p9:=textplot([F2[1]+0.2,F2[2]+0.2,"F[2]"],align=ABOVE):
> display(seq(p||i,i=1..9));
```



Парабола: $|\mathbf{z}-\mathbf{I}|=\mathbf{Re}(\mathbf{z}-\mathbf{2I})+\mathbf{1}$

```
> `|z-I|=Re(z-2*I)+1`;
```

$$|z-I|=Re(z-2I)+1$$

Найдем образ данного соотношения.

```
> with(plots): with(plottools): with(geometry): z:=x+y*I:
> implicitplot(abs(z-I)=Re(z-2*I)+1, x=-5..5,y=-5..5,
  scaling=unconstrained,color=black, thickness=4);
```



Видим, что получилась парабола.

Замечание. Параболой называется геометрическое место точек, для каждой из которых расстояние до некоторой фиксированной точки плоскости, называемой фокусом, равно расстоянию до некоторой фиксированной прямой, называемой директрисой (предполагается, что эта прямая не проходит через фокус).

Чтобы убедится в правильности наших выводов, получим каноническое уравнение этой параболы.

```
> evalc(abs(z-I)=Re(z-2*I)+1); evalc(abs(z-I)^2=(Re(z-2*I)+1)^2);
> expand(%); lhs(%)-rhs(%)=0; y^2=solve(%,y^2);
```

$$\sqrt{x^2+y^2-2y+1}=x+1$$

$$x^2+y^2-2y+1=(x+1)^2$$

$$x^2+y^2-2y+1=x^2+2x+1$$

$$y^2-2y-2x=0$$

$$y^2=2y+2x$$

С помощью функции conic получим все ее характеристики.

```
> eqn:=lhs(%%): conic(c1,eqn=0,[x,y]): detail(c1);
```

name of the object: c1
form of the object: parabola2d
vertex: [-1/2, 1]
focus:$[0,1]$

directrix: $x + 1 = 0$
equation of the parabola: $y^2 - 2 * y - 2 * x = 0$

Еще раз строим параболу, ее директрису и фокус.

```
> p2:=implicitplot(Equation(directrix_c1),x=-4..4,y=-4..4,):
> p3:=implicitplot(Equation(c1),x=-4..4,y=-4..4,thickness=4):
> zx:=1.5: zy:=solve(subs(x=zx,Equation(c1)),y)[1]:
> v1:=arrow(coordinates(focus_c1),[zx,zy], 0.01, 0.07, 0.3):
> v2:=arrow([-1,zy],[zx,zy], 0.007, 0.07, 0.3):
> p4:=pointplot(coordinates(focus_c1),symbol=CIRCLE,
  symbolsize=14):
> p5:=pointplot([zx,zy],symbol=CIRCLE,symbolsize=14):
> p6:=pointplot([-1,zy],symbol=CIRCLE,symbolsize=14):
> p7:=textplot([0+0.1,1-0.2,"z0"],align=RIGHT):
> p8:=textplot([zx+0.1,zy+0.4,"z"],align=RIGHT):
> display(seq(p||i,i=2..8),v1,v2,scaling=unconstrained);
```



Внутренность угла: $\pi/\mathbf{6} \leq \arg(\mathbf{z} - \mathbf{z_0}) < \pi/\mathbf{4}$

```
> `Pi/6<=arg(z-z0)<Pi/4`;
```

$$\pi/6 \leq \arg(z - z0) < \pi/4$$

```
> z0:=0.5+0.5*I; x0:=Re(z0): y0:=Im(z0): alpha:=Pi/6; beta:=Pi/4;
```

$$z0 := 0.5 + 0.5I$$
$$\alpha := \frac{\pi}{6}$$
$$\beta := \frac{\pi}{4}$$

Изобразим данную область, используя функцию inequal, предназначенную для рисования областей заданных системой *линейных* неравенств.

```
> with(plots): with(plottools):
> inequal(y>alpha*(x-x0)+y0, y<beta*(x-x0)+y0, x=x0, y=y0,
```

```
x=0..2,y=0..2, optionsfeasible=(color=white),
optionsopen=(color=black,thickness=2,linestyle=3),
optionsclosed=(color=black,thickness=2,linestyle=1),
optionsexcluded=(color=grey));
```



Полуплоскость: $\mathbf{Re(iz + 3i) \le -1}$

```
> `Re(I*z+3*I)`<=-1; Re(I*z+3*I)<=-1:
```

$$\mathrm{Re}(I * z + 3 * I) \le -1$$

```
> evalc(subs(z=x+I*y,Re(I*z+3*I)<=-1));
```

$$-y <= -1$$

```
> with(plots):
> inequal(evalc(subs(z=x+I*y,Re(I*z+3*I)<=-1)),x=-5..5,y=-5..5,
  optionsfeasible=(color=white),
  optionsopen=(color=black,thickness=2,linestyle=3),
  optionsclosed=(color=black,thickness=2,linestyle=1),
  optionsexcluded=(color=grey));
```



Серединный перпендикуляр: $|\mathbf{z}-\mathbf{z1}|=|\mathbf{z}-\mathbf{z2}|$

```
> abs(z-z1)=abs(z-z2);
```

$$|z-z1|=|z-z2|$$

```
> z1:=-1-I; z2:=-2+I; eq:=abs(z-z1)=abs(z-z2): eq;
```

$$z1:=-1-I$$
$$z2:=-2+I$$
$$|z+1+I|=|z+2-I|$$

```
> eq2:=evalc(subs(z=x+I*y,eq)): eq2;
```

$$\sqrt{x^2+2x+2+y^2+2y}=\sqrt{x^2+4x+5+y^2-2y}$$

```
> y:=solve(eq2,y); y1:=subs(x=2,y):
```

$$y:=\frac{x}{2}+\frac{3}{4}$$

```
> with(plots): p1:=plot(y,x=-3.5..3,color=black,thickness=4):
> p2:=plot([-1,-1],[-2,1],color=black):
> A:=[Re(z1),Im(z1)]: B:=[Re(z2),Im(z2)]: C:=[2,y1]:
> p3:=plottools[polygon]([A,B,C,A],linestyle=0,thickness=1):
> p4:=textplot([Re(z1)+0.2,Im(z1),"z1"],align=RIGHT):
> p5:=textplot([Re(z2)-0.2,Im(z2),"z2"],align=LEFT):
> p6:=textplot([2+0.2,y1-0.1,"z"],align=LEFT):
> p7:=pointplot([Re(z1),Im(z1)],color=black,symbol=CIRCLE,
  symbolsize=14):
> p8:=pointplot([Re(z2),Im(z2)],color=black,symbol=CIRCLE,
  symbolsize=14):
> p9:=pointplot([2,y1],color=black,symbol=CIRCLE,symbolsize=14):
> display([seq(p||i,i=1..9)],scaling=constrained);
```



Множество точек z, удовлетворяющих уравнению $|z - z_1| = |z - z_2|$ — это серединный перпендикуляр отрезка, соединяющего точки z_1 и z_2, т.е. геометрическое место точек плоскости, равноудаленных от этих точек.

Следовательно, это уравнение прямой, перпендикулярной отрезку, соединяющему точки z_1, z_2, и проведенной через его середину.

Пример 7. *Описать с помощью уравнений и неравенств множество, изображенное на рисунке (дополнение области заштриховано).*



Чтобы получить неравенства, определяющие множества, сначала нужно составить уравнения, описывающие границы.

Геометрически заданное множество есть "внутренность" одной ветви гиперболы $x^2 - y^2 = 2$ с разрезом (выброшенным лучом – частью вещественной оси от 2 до $+\infty$). Итак, данное множество можно описать

$$\left\{ \begin{array}{ll} \left\{ \begin{array}{l} x^2 - y^2 \geq 2, \\ x \geq \sqrt{2}, \end{array} \right. , & \text{при } y \neq 0; \\ \sqrt{2} \leq x \leq 2, & \text{при } y = 0. \end{array} \right.$$

Или выразив x в уравнении гиперболы:

$$\left\{ \begin{array}{ll} x \geq \sqrt{2 + y^2}, & \text{при } y \neq 0; \\ \sqrt{2} \leq x \leq 2, & \text{при } y = 0. \end{array} \right.$$

Запишем эти ограничения в комплексной форме, для этого вспомним, что

$$x = \operatorname{Re} z = \frac{z + \overline{z}}{2} \quad y = \operatorname{Im} z = \frac{z - \overline{z}}{2i}.$$

Таким образом:

$$\left\{ \begin{array}{ll} \operatorname{Re} z \geq \sqrt{2 + (\operatorname{Im} z)^2}, & \text{при } \operatorname{Im} z \neq 0; \\ \sqrt{2} \leq \operatorname{Re} z \leq 2, & \text{при } \operatorname{Im} z = 0. \end{array} \right.$$

Обратите внимание, что имеется много других способов описать это множество, приведен лишь один из возможных вариантов.

Осталось это множество изобразить.

```
> with(plots): eqn:=(Re(z))^2-(Im(z))^2-2: eqn>=0;
> z:=x+y*I: eqn:=evalc(eqn): eqn<0;
```

$$0 <= \mathrm{Re}(z)^2 - \mathrm{Im}(z)^2 - 2$$

$$x^2 - y^2 < 2$$

```
> p1:=implicitplot(eqn>=0,x=0..4,y=-3..3, filled=true,
  coloring=[white,grey],linestyle=1,thickness=3):
> p3:=pointplot([2,0],symbol=CIRCLE,symbolsize=10):
> p4:=pointplot([2,0],symbol=CIRCLE,symbolsize=18):
> p2:=plottools[line]([2,0],[4,0],linestyle=1,thickness=3):
> display(seq(p||i,i=1..4),scaling=CONSTRAINED);
```



2.2 Задачи для самостоятельного решения

1. Указать и изобразить линии, определяемые условиями:

1) $\mathrm{Im}\, z^2 = 2$,

2) $\mathrm{Re}(\overline{z^2 - \overline{z}}) = 3 - \mathrm{Im}\, z$,

3) $\mathrm{Re}(2 + i\overline{z}) = |z|$,

4) $z^2 + \overline{z}^2 = 1$,

5) $z\overline{z} + (z - \overline{z})i = 2$,

6) $\mathrm{Re}\overline{z}^2 = 1$,

7) $|z| - Re(z) = 12$,

8) $|z - i| + |z + i| = 4$,

9) $\mathrm{Im}(\overline{z^2 - \overline{z}}) = 2 - \mathrm{Im}\, z$,

10) $\mathrm{Re}\, z^2 = 3$,

11) $2z\overline{z} + (2 - i)z + (2 + i)\overline{z} = 2$,

12) $\mathrm{Re}(z^2 - \overline{z}) = 0$,

13) $\mathrm{Im}(1 + z) = |z|$,

14) $\mathrm{Im}(\overline{z}^2 - z) = 1 - 2\mathrm{Re}\, z$,

15) $|z + 2i| = |z - 2|$,

16) $|z| - 3\mathrm{Im}\, z = 6$.

2. Выяснить геометрический смысл соотношения и изобразить.

1) $\operatorname{Re}(iz-1) \le 6$,	2) $\operatorname{Im}(iz-1) > -6$,
3) $\lvert z-2i\rvert > \lvert z+2\rvert$,	4) $0 \le \arg(2z+2i) < \frac{\pi}{2}$
5) $\operatorname{Re}(iz-2) \ge 6$,	6) $\operatorname{Im}(z^2+iz-2) > -6$,
7) $\lvert z+2i\rvert \le \lvert z-2+i\rvert$,	8) $0 \le \arg(z-i) < \frac{2\pi}{3}$
9) $\operatorname{Re}(z^2-3z+2i) \le 4$,	10) $\lvert z+4i\rvert \ge \lvert zi\rvert$,
11) $\frac{\pi}{2} \le \arg(z+1-i) < \frac{3\pi}{4}$,	12) $\operatorname{Im}(3z+2i) < -4$,
13) $\operatorname{Re}(3z-i) \le 2$,	14) $0 < \arg(z+i) < \frac{3\pi}{4}$,
15) $\lvert z+2i\rvert > \lvert z-2i\rvert$,	16) $\operatorname{Im}(z^2+3z-i) < -2$.

3. Описать с помощью уравнений и неравенств множество, изображенное на рисунке (дополнение области заштриховано).

1)


2)


3)


4)


5)


6)


7)



8)



9)



10)



11)



12)



13)



14)



15)



16)



3 Элементарные функции

Пример 8. *Найти значения основных элементарных функций от $z = 2+i$.*

Для каждой из элементарных функций будем выполнять одну и туже последовательность действий:

- вспомним определение,
- вычислим значение функции в точке $z = 2 + i$ с помощью встроенных функций Maple,
- продолжим вычисления, если необходимо получить численное значение данного выражения.

3.1 Экспонента e^z

Сначала вспомним определение экспоненты:

```
>  z:=x+y*I; e^'z'=evalc(exp(z));
```

$$z := x + yI$$

$$e^z = e^x \cos(y) + e^x \sin(y)I$$

Вычислим значение экспоненты в точке $2+i$ с помощью встроенных функций

```
> e^(2+I)=evalc(exp(2+I));
```

$$e^{(2+I)} = e^2 \cos(1) + e^2 \sin(1)I$$

Продолжим вычисления, чтобы получить численное значение данного выражения

```
> e^(2+I)=evalf(evalc(exp(2+I)));
```

$$e^{(2+I)} = 3.992324049 + 6.217676312\ I$$

3.2 Вычисление значений тригонометрических и гиперболических функций комплексной переменной

Как известно из курса комплексного анализа, тригонометрические и гиперболические функции комплексной переменной определяются через экспоненту и являются <u>однозначными</u>.

Для примера подробнее рассмотрим лишь *синус,* все остальные находятся аналогично.

3.2.1 Синус $\sin z$

Вспомним известное из курса комплексного анализа, определение синуса:

```
> sin(z)=(exp(z*I)-exp(-z*I))/(2*I);
```

$$\sin(z) = -\frac{1}{2}I\left(e^{(zI)} - e^{(-zI)}\right)$$

Обратите внимание на то, что Maple выносит I за скобочки и из знаменателя переносит его в числитель. Почти такое же выражение получается, если воспользоваться функциями `convert` и `sin`.

```
> sin(z)=convert(sin(z));
```

$$\sin(z) = -\frac{1}{2}I\left(e^{(zI)} - \frac{1}{e^{(Iz)}}\right)$$

Для вычислений, однако, удобнее выражение через вещественные функции:

```
> sin(z)=evalc(sin(x+y*I));
```

$$\sin(z) = \sin(x)\cosh(y) + \cos(x)\sinh(y)I$$

Далее вычислим значение синуса в точке $2+i$:

```
> sin(2+I)=evalc(sin(2+I));
```

$$\sin(2+I) = \sin(2)\cosh(1) + \cos(2)\sinh(1)I$$

Получим численное значение этого выражения:

```
> sin(2+I)=evalf(evalc(sin(2+I)));
```

$$\sin(2+I) = 1.403119251 - .4890562591I$$

Обратите внимание, что англоязычной литературе и в пакете Maple также стандартные обозначения некоторых функций отличаются от принятых у нас:

тангенс	$\mathrm{tg}\, z$	tan(z)
котангенс	$\mathrm{ctg}\, z$	cot(z)
синус гиперболический	$\mathrm{sh}\, z$	sinh(z)
косинус гиперболический	$\mathrm{ch}\, z$	cosh(z)
тангенс гиперболический	$\mathrm{th}\, z$	tanh(z)
котангенс гиперболический	$\mathrm{cth}\, z$	coth(z).

Аналогично обозначаются обратные тригонометрические и гиперболические функции:

$$\mathrm{arctan(z)}, \mathrm{arccot(z)}, \mathrm{arcsinh(z)}, \mathrm{arccosh(z)}, \mathrm{arctanh(z)}, \mathrm{arccoth(z)}.$$

3.3 Вычисление логарифма комплексного числа $\mathrm{Ln}\, z$

Используя функцию `ln(z)`, получим лишь главное значение многозначной функции $\mathrm{Ln}\, z$. Поэтому определим сами нужную нам многозначную функцию $\mathrm{Ln}\, z$ и выполним все необходимые вычисления с ее помощью. Ее же будем использовать в дальнейшем при нахождении других многозначных элементарных функций.

```
> Ln:=(z)->ln(z)+2*Pi*k*I: `Ln(z)`=Ln(z);
```

$$Ln(z) = \ln(z) + 2\pi k\ I$$

В этой части всюду под k подразумевается произвольное целое число. Теперь вспомним необходимые определения:

```
> z:=x+y*I; `Ln(z)`=Ln(z);
> `Ln(z)`=evalc(Ln(z));
```

$$z := x + yI$$

$$Ln(z) = \ln(x + yI) + 2I\pi k$$

$$Ln(z) = \frac{1}{2}\ln(x^2 + y^2) + (\arctan(y, x) + 2\pi k)I$$

Обратите внимание, здесь использована встроенная функция `arctan(y,x)`, это — не совсем арктангенс! Хотя, после вычислений она и превращается в настоящий арктангенс.

Далее вычисляем значение функции в точке $z = 2 + i$.

```
> `Ln(2+I)`=evalc(Ln(2+I));
```

$$Ln(2 + I) = 1/2\ln(5) + (\arctan(1/2) + 2\pi k)I$$

Находим численное значение данного выражения:

```
> `Ln(2+I)`=evalf(evalc(Ln(2+I)));
```

$$Ln(2 + I) = 0.8047189560 + 1.I(0.4636476090 + 6.283185308k)$$

3.4 Вычисление значений обратных тригонометрических и гиперболических функций комплексной переменной

Обратные тригонометрические и гиперболические функции, показательная a^z и степенная функций z^a комплексной переменной, определяются через логарифм и являются многозначными.

Для примера подробнее рассмотрим *арксинус.*

3.4.1 Арксинус $\operatorname{Arcsin} z$

Арксинус определяется как обратная функция к синусу. Из-за периодичности синуса получается многозначность арксинуса, как функции комплексной переменной. Синус выражается через экспоненту, следовательно, арксинус – через логарифм. Аналогично, встроенная функция `arcsin(z)` дает лишь главное значение.

Итак, следуя выбранной схеме, вычислим $\operatorname{Arcsin}(2+i)$.
Запишем развернутое выражение через логарифм:

```
> Arcsin:=(z)->arcsin(z)+2*Pi*k: `Arcsin(z)`=Arcsin(z);
```

$$Arcsin(z) = arcsin(z) + 2\pi k$$

```
> `Arcsin(z)`=convert(Arcsin(z),ln);
```

$$Arcsin(z) = -\ln(\sqrt{1-z^2} + zI)I + 2\pi k.$$

Вычисление значения функции в точке $z = 2 + i$:

```
> `Arcsin(2+I)`=evalc(Arcsin(2+I));
```

$$Arcsin(2+I) =$$
$$\arcsin\left(\frac{\sqrt{10}}{2} - \frac{\sqrt{2}}{2}\right) + \left(\ln\left(\frac{\sqrt{10}}{2} + \frac{\sqrt{2}}{2} + \sqrt{\left(\frac{\sqrt{10}}{2} + \frac{\sqrt{2}}{2}\right)^2 - 1}\right) - 2\pi k\ I\right) I$$

```
> `Arcsin(2+I)`=evalf(evalc(Arcsin(2+I)));
```

$$Arcsin(2+I) = 1.063440024 + 1.I(1.469351744 - 6.283185308kI)$$

Пример 9. *Найти все корни уравнения* $\sin z = 10$.

Заметим, что вещественных корней данное уравнение не имеет! Попытаемся его решить, не задумываясь, сразу:

```
> sin(z)=10; z=solve(%,z);
```

$$\sin(z) = 10$$

$$z = \arcsin(10)$$

Почти получилось, осталось лишь вспомнить, что решением этого уравнения будет многозначная функция арксинус (см. пункт 3.4.1).

```
> Arcsin:=(z)->arcsin(z)+2*Pi*k;
```

$$Arcsin := z \longrightarrow arcsin(z) + 2\pi k$$

```
> `Arcsin(z)`=convert(Arcsin(z),ln);
```

$$Arcsin(z) = -\ln(\sqrt{1-z^2} + zI)I + 2\pi k.$$

Вычислим значения функции в точке 10 :

```
> `Arcsin(10)`=evalc(Arcsin(10));
```

$$Arcsin(10) = \frac{\pi}{2} + \left(-\ln\left(10 + 3\sqrt{11}\right) - 2\pi k I\right) I$$

Не забываем, какое значение аргумента комплексного числа выбрано в Maple главным! (см. сноску на стр. 5)

```
> `Arcsin(10)`=evalf(evalc(Arcsin(10)));
```

$$Arcsin(10) = 1.570796327 + 1.I(-2.993222846 - 6.283185308kI)$$

Подробнее о решении такого типа уравнений можно прочесть, например, в [11, пример 1.24.].

3.5 Задачи для самостоятельного решения

1. Вычислить значение однозначной функции.

1) $\cos(2-2i)$,	2) $\mathrm{tg}(1-2i)$,	3) $\mathrm{ctg}(1+3i)$,	4) $\mathrm{sh}(2+2i)$,
5) $\mathrm{ch}(i-2)$,	6) $\mathrm{th}(i-1)$,	7) $\mathrm{cth}(3+i)$,	8) $\cos(4+i)$,
9) $\mathrm{tg}(3+2i)$,	10) $\mathrm{ctg}(3-2i)$,	11) $\mathrm{sh}(-1-2i)$,	12) $\mathrm{ch}(1-i)$,
13) $\mathrm{th}(3-i)$,	14) $\mathrm{cth}(3i-2)$,	15) $\mathrm{tg}(4+3i)$,	16) $\mathrm{ctg}(5+i)$.

2. Вычислить значение многозначной функции.

1) $(2+i)^{(4+3i)}$,	2) $\mathrm{Arccos}(7-i)$,	3) $\mathrm{Arctg}(1-2i)$,	4) $\mathrm{Arcctg}(5-i)$,
5) $\mathrm{Arcsh}(2+2i)$,	6) $\mathrm{Arcch}(i-2)$,	7) $\mathrm{Arcth}(i-1)$,	8) $\mathrm{Arccth}(3+i)$
9) $\mathrm{Arccos}(4+i)$,	10) $\mathrm{Arctg}(3+2i)$,	11) $\mathrm{Arcctg}(3-2i)$,	12) $\mathrm{Arcsh}(1+i)$,
13) $\mathrm{Arcch}(1-i)$,	14) $\mathrm{Arcth}(3-i)$,	15) $\mathrm{Arccth}(3i-2)$,	16) $(2-i)^{(5+i)}$.

3. Найти все корни уравнения.

1) $\cos z = 2i$,	2) $\mathrm{tg} z = -2+i$,	3) $\mathrm{ctg} z = -2i$,	4) $\mathrm{sh}\, z = 2$,
5) $\mathrm{ch}\, z = 2-i$,	6) $\mathrm{th}\ z = i$,	7) $\mathrm{cth}\ z = 2+i$,	8) $\cos z = 1+i$,
9) $\mathrm{tg} z = -1+i$,	10) $\mathrm{ctg} z = -2+i$,	11) $\mathrm{sh}\, z = 4$,	12) $\mathrm{ch}\, z = 1-i$,
13) $\mathrm{th}\ z = 4i$,	14) $\mathrm{cth}\ z = 1+i$,	15) $\cos z = 5i$,	16) $\mathrm{tg} z = -2-i$.

4 Дифференцирование

4.1 Исследование дифференцируемости функции, условия Коши-Римана, вычисление производной

Пример 10. *Определить, существует ли область D, в которой функция $f(z) = |z|^2$ дифференцируема; найти производную $f'(z)$ при $z \in D$.*

Вам, должно быть, хорошо известна

Теорема. *Для того, чтобы функция $w = f(z) = u(x, y) + iv(x, y)$ была дифференцируема в точке $z_0 = x_0 + iy_0$*

$\Leftrightarrow$

1) существовали непрерывные частные производные $u(x, y)$ и $v(x, y)$;

2) выполнялись в точке z_0 ***условия Коши-Римана:*** $\begin{cases} u'_x &= v'_y \\ u'_y &= -v'_x \end{cases}$.

Следовательно, для исследования функции на дифференцируемость и нахождения ее производной следует выполнить следующие операции.

1) Для заданной функции $f(z)$ найти действительную и мнимую части:

$$u = u(x, y) = \mathrm{Re}(f(z)), \quad v = v(x, y) = \mathrm{Im}(f(z)).$$

2) Найти частные производные функций $u(x, y)$, $v(x, y)$.

3) Проверить выполнение условий Коши-Римана. Точки, в которых эти условия не выполняются, являются точками, где функция не дифференцируема. Точки, в которых условия выполняются и частные производные являются непрерывными, принадлежат искомой области D, где функция дифференцируема.

4) Записать выражение производной в точках дифференцируемости по одной из формул:

$$f'(z) = \frac{\partial u}{\partial x} + i\frac{\partial v}{\partial x}, \qquad f'(z) = \frac{\partial v}{\partial y} - i\frac{\partial u}{\partial y},$$

$$f'(z) = \frac{\partial u}{\partial x} - i\frac{\partial u}{\partial y}, \qquad f'(z) = \frac{\partial v}{\partial y} + i\frac{\partial u}{\partial x}.$$

1) Определим функцию и найдем ее действительную и мнимую части:

```
> f:=z->abs(z)*abs(z): `f(z)`=f(z);
```

$$f(z) = |z|^2$$

```
> z:=x+y*I: `f(z)`:=evalc(f(z));
```

$$f(z) := x^2 + y^2$$

```
> u:=(x,y)->Re(f(z)): `u(x,y)`:=evalc(u(x,y));
> v:=(x,y)->Im(f(z)): `v(x,y)`:=evalc(v(x,y));
```

$$u(x, y) := x^2 + y^2$$

$$v(x, y) := 0$$

2) Найдем частные производные функций $u(x, y),\ v(x, y):$

```
> `u′`[x]:=diff(`u(x,y)`,x); `u′`[y]:=diff(`u(x,y)`,y);
> `v′`[x]:=diff(`v(x,y)`,x); `v′`[y]:=diff(`v(x,y)`,y);
```

$$u'_x := 2x$$

$$u'_y := 2y$$

$$v'_x := 0$$

$$v'_y := 0$$

3) Проверим выполнение условий Коши-Римана:

```
> `u′`[x]=`v′`[y];
> `u′`[y]=-`v′`[x];
```

$$u'_x = v'_y$$

$$u'_y = -v'_x$$

```
> `u′`[x]=`v′`[y];
> `u′`[y]=-`v′`[x];
```

$$2x = 0$$

$$2y = 0$$

```
> sys:=`u′`[x]=`v′`[y],`u′`[y]=-`v′`[x]:
> sol:=solve(sys,x,y);
```

$$sol := \{\{x = 0, y = 0\}\}$$

Выполнение условий Коши-Римана является необходимым условием дифференцируемости функции $f(z)$ в точке. Следовательно, их невыполнения достаточно для утверждения о том, что функция не является дифференцируемой в соответствующей точке. Таким образом, если система не имеет решений, то функция нигде не дифференцируема.

Условия Коши-Римана не являются достаточными. В соответствующей точке должны быть дифференцируемы функции $u(x, y)$ и $v(x, y)$. Напомним, что условием дифференцируемости функции двух действительных переменных в точке является существование и непрерывность частных производных в этой точке. Не забывайте, что Maple не умеет проверять непрерывность!

```
> z:=′z′:
> if sol={ } then
> `Условия Коши-Римана не выполнены ни в одной точке плоскости`;
> else
```

```
> `Условия Коши-Римана выполнены на множестве D1`=sol;
> `f'(z)`:=`u′`[x]+I*`v′`[x];
> `f'(z)`:=simplify(subs(y=-I/2*(z-conjugate(z)),
> subs(x=1/2*(z+conjugate(z)),`u′`[x]+I*`v′`[x])));
> end if;
```

Условия Коши-Римана выполнены на множестве $D1 = \{\{x = 0, y = 0\}\}$

$$f'(z) := 2x$$

$$f'(z) := 2\mathrm{Re}(z)$$

Обратите внимание, что D1 это всего лишь множество, где выполнены условия Коши-Римана. Оно же является множеством D, на котором дифференцируема функция, если частные производные $u(x, y)$ и $v(x, y)$ на нем непрерывны. В данном случае очевидно, что все частные производные функций $u(x, y)$ и $v(x, y)$ непрерывны во всей полоскости, в том числе и в точке 0.

Функция $f(z)$ дифференцируема лишь в точке $z = 0$, ее производная

$$f'(z) = 2\ \mathrm{Re}(z) = 2\ \mathrm{Re}(0) = 0.$$

4.2 Геометрический смысл производной

Производная $f'(z)$, как функция комплексной переменной, определяет отображение области D, области дифференцируемости функции $f(z)$ в область G. В каждой точке $z_0 \in D$ определено комплексное число $f'(z_0)$, следовательно, определены $|f'(z_0)|$ и $\arg f'(z_0)$, если $f'(z_0) \neq 0$. Геометрически – число $|f'(z_0)|$ – длина радиуса-вектора, а $\arg f'(z_0)$ - угол наклона радиуса-вектора точки $f'(z_0)$, к действительной оси.

Возникает вопрос: как характеризуют эти величины само отображение $w = f(z)$ в точке z_0. Для функции действительной переменной аналогичный вопрос решается просто: производная $f'(x_0)$ определяет угловой коэффициент касательной, проведенной к кривой $y = f(x)$ в точке $(x_0, f(x_0))$.

Геометрический смысл

• **модуля производной** $|f'(z_0)|$ состоит в том, что он равен коэффициенту линейного растяжения (сжатия) бесконечно малых векторов в точке z_0 :

$$|f'(z_0)| = k;$$

• **аргумента производной** $arg\ \ f'(z_0)$ состоит в том, что он равен углу поворота бесконечно малых векторов в точке z_0 :

$$\arg\ f'(z_0) = \theta.$$

Подробнее с геометрическим смыслом производной функции комплексной переменной можно ознакомится, например, в [6-10].

Замечание: серым цветом залита та часть плоскости, которая сжимается.

Пример 11. *Какая часть плоскости растягивается (сжимается) при отображении, осуществляемом функцией $f(z) = -2/z$. Найти также коэффициент растяжения k и угол поворота θ в точке $z_0 = 1$.*

Аналитическое решение задач такого типа рассматриваются, например, в [11], примеры 3.3–3.6. Поэтому решение данного примера приводится с минимальными комментариями.

Определим функцию и точку, заданные по условию:

```
> f:=z->-2/z: `f(z)`=f(z); z0:=1;
```

$$f(z) = -\frac{2}{z}$$

$$z0 := 1$$

Найдем коэффициент растяжения и угол поворота в точке z_0.

```
> w:=z->diff(f(z),z): `f′(z)`:=w(z);
> k=abs(`f′(z0)`); k:=abs(subs(z=z0,w(z)));
> if k=1 then
> `k=1 - в окрестности точки z0 растяжения(сжатия) нет`
> elif evalf(k)<1 then
> `k<1 - в окрестности точки z0 сжатие`;
> else
> `k>1 - в окрестности точки z0 растяжение`;
> end if;
```

$$f\,'(z) := \frac{2}{z^2}$$

$$f'(z0) = 2$$

$$k = |f'(z0)|$$

$$k := 2$$

k>1 - в окрестности точки z0 растяжение

```
> theta:=arg(`f'(z0)`);
> theta:=arg(‘f'(z0)‘);
> theta:=argument(subs(z=z0,w(z)));
> if theta=0 then `в окрестности точки z0 поворота нет`
> else `в окрестности точки z0 угол поворота theta`;
> end if;
```

$$\theta := \arg(f'(z0))$$

$$\theta := 0$$

в окрестности точки z0 поворота нет

Найдем ту часть плоскости, которая при отображении $f(z)$ сжимается, обозначим его Ω и на рисунке зальем серым.

```
> z:='z': `f'(z)`=w(z);
> abs(`f'(z)`)<1;
> abs(w(z))<1;
> z:=x+I*y: Omega:={evalc(%%)};
```

$$f'(z) := \frac{2}{z^2}$$

$$|f'(z)| < 1$$

$$\frac{2}{z^2} < 1$$

$$\Omega := \{\frac{2}{x^2+y^2} < 1\}$$

Используя стандартные возможности Maple, изобразим множество Ω.

```
> with(plots):
> p1:=implicitplot(Omega,x=-2..2,y=-2..2,scaling=CONSTRAINED,
  filled=true, linestyle=3,thickness=2,coloring=[grey,white]):
> p2:=textplot([-0.5,0.5,"растяжение"],align=RIGHT,color=blue):
> p3:=textplot([-1,1.5,"сжатие"],align=ABOVE,color=blue):
> p4:=pointplot([Re(z0),Im(z0)],color=black,symbol=CIRCLE,
  symbolsize=14):
> p5:=textplot([Re(z0),Im(z0)+0.1,"z0"],align=ABOVE):
> display(p1,p2,p3,p4,p5);
```



4.3 Связь аналитических функций с гармоническими

Пример 12. *Найти аналитическую функцию $f(z)$ по изветной ее действительной части $u(x,y) = 2e^x\cos(y)$ при условии $f(0) = 2+2i$.*

Зададим условия задачи, определив действительную часть аналитической функции $f(z)$ как функцию $u(x,y)$ и присвоив константам $z0$ и $f0$ соответствующие условию значения:

```
> u:=(x,y)->2*exp(x)*cos(y): `u(x,y)`=u(x,y);
> z0:=0; f0:=2+2*I;
```

$$u(x,y) = 2e^x\cos(y)$$

$$z0 := 0$$

$$f0 := 2+2I$$

Напомним себе условия Коши-Римана.

```
> `u'`[x]=`v'`[y];
> `u'`[y]=-`v'`[x];
```

$$u'_x = v'_y$$
$$u'_y = -v'_x$$

Из условий Коши-Римана определим частные производные искомой мнимой части функции $f(z)$.

```
> diff(v(x,y),x)=-diff(u(x,y),y);
> diff(v(x,y),y)=diff(u(x,y),x);
```

$$\frac{\partial}{\partial x}v(x,y) = 2e^x\sin(y)$$

$$\frac{\partial}{\partial y}v(x,y) = 2e^x\cos(y)$$

В курсе комплексного анализа были рассмотрены два простые способа нахождения неизвестной мнимой части $v(x,y)$, один из которых основан на вычислении криволинейного интеграла:

$$v(x,y) = \int\limits_{(x_0,y_0)}^{(x,y)} dv = \int\limits_{(x_0,y_0)}^{(x,y)} \frac{\partial v}{\partial x}dx + \frac{\partial v}{\partial y}dy + c = \int\limits_{(x_0,y_0)}^{(x,y)} -\frac{\partial u}{\partial y}dx + \frac{\partial u}{\partial x}dy + c.$$

Причем всегда удобнее найти функцию $v(x,y)$ с точностью до постоянного слагаемого, и лишь в самом конце подобрать константу .

```
> v:=(x,y)->evalc(subs(t=x,subs(y=0,int(-diff(u(t,y),y),t=0..x)))
  +int(diff(u(x,t),x),t=0..y):\‘v(x,y)\‘=v(x,y);
```

$$v(x,y) = 2e^x\sin(y)$$

Вычислим частные производные найденной функции, чтобы убедится в правильности вычислений.

```
> `diff(v(x,y),x)`=diff(v(x,y),x); `diff(v(x,y),y)`=diff(v(x,y),y);
```

$$\frac{\partial}{\partial x} v(x, y) = 2e^x \sin(y)$$

$$\frac{\partial}{\partial y} v(x, y) = 2e^x \cos(y)$$

Таким образом найдена функция $f1(x, y)$ двух веществественных переменных равная искомой с точность до константы.

```
> f1:=(x,y)->u(x,y)+I*v(x,y): `f1(x,y)`=f1(x,y);
```

$$f1(x, y) = 2e^x \cos(y) + 2Ie^x \sin(y)$$

Как известно из свойств комплексных чисел:

$$x = \operatorname{Re} z = \frac{z + \overline{z}}{2} \quad y = \operatorname{Im} z = \frac{z - \overline{z}}{2i}$$

Подставим эти тождества в определение функция $f1(x, y)$.

```
> z:='z':
> f2:=(z)->simplify(evalc(subs(y=-I/2*(z-conjugate(z)),
  subs(x=1/2*(z+conjugate(z)),f1(x,y))))): `f2(z)`=f2(z);
```

$$f2(z) = 2e^z$$

Теперь найдем подходящую константу c так, чтобы функция $f(z) = f2(z) + c$ удовлетворяла условию $f(z0) = f0$.

```
> c:='c': evalc(f2(z0))+c=f0;
```

$$2 + c = 2. + 2.I$$

```
> c:=fsolve(evalc(f2(z0))+c=f0,c,complex);
```

$$c := 2.I$$

Таким образом найден ответ:

```
> `f(z)`=f2(z)+c;
```

$$f(z) = 2e^z + 2.I$$

Проверка:

```
> `u(x,y)=Re(f(z))`=evalc(Re((f2(x+y*I)+c)));
> `f0=f(z0)`=f2(z0)+c;
```

$$u(x, y) = Re(f(z)) = 2e^x \cos(y)$$

$$f0 = f(z0) = 2. + 2.I$$

4.4 Задачи для самостоятельного решения

1. Определить, существует ли область D, в которой функция $f(z)$ дифференцируема; найти производную $f'(z)$ при $z \in D$.

1)	$f(z) = \operatorname{Im} z$	5)	$f(z) = z \operatorname{Im} z$	9)	$f(z) = x^2y^2$	13)	$f(z) = e^z - \frac{5}{z}$
2)	$f(z) = \sin z^2$	6)	$f(z) = \cos z^2$	10)	$f(z) = \operatorname{sh} z^2$	14)	$f(z) = z^2 - 3iz$
3)	$f(z) = \overline{z} \operatorname{Re} z$	7)	$f(z) = z + 2i\overline{z}$	11)	$f(z) = \lvert z\rvert\overline{z}$	15)	$f(z) = \lvert z\rvert \operatorname{Im} z$
4)	$f(z) = \overline{z} \operatorname{Im} z$	8)	$f(z) = 2\overline{z} + z^2$	12)	$f(z) = \operatorname{ch} z^2$	16)	$f(z) = i(x^2 - y^2)$

2. Какая часть плоскости растягивается (сжимается) при отображении, осуществляемом функцией $f(z)$. Найти также коэффициент растяжения k и угол поворота θ в точке z_0.

1)	$f(z) = \frac{z-i}{z+i}$,	$z_0 = 1$	2)	$f(z) = \ln(-iz)$,	$z_0 = 2$
3)	$f(z) = z^4 + 3i$,	$z_0 = -1$	4)	$f(z) = e^{-z}$,	$z_0 = i$
5)	$f(z) = \frac{z-i}{z+1}$,	$z_0 = 2i$	6)	$f(z) = \frac{2}{z}$,	$z_0 = -1$
7)	$f(z) = z^3 + 4i$,	$z_0 = -1$	8)	$f(z) = e^{z-2}$,	$z_0 = -i$
9)	$f(z) = \frac{2i}{z}$,	$z_0 = i$	10)	$f(z) = \ln(z-1)$,	$z_0 = i$
11)	$f(z) = z^3 - 4i$,	$z_0 = i$	12)	$f(z) = e^{iz}$,	$z_0 = -i$
13)	$f(z) = \frac{z+i}{z-i}$,	$z_0 = 0$	14)	$f(z) = \ln z$,	$z_0 = 2i$
15)	$f(z) = z^4 - 3i$,	$z_0 = i$	16)	$f(z) = \frac{z+i}{z+1}$,	$z_0 = i$

3. Найти аналитическую функцию $f(z)$ по изветной ее

1) вещественной части $u(x,y) = 2\sin\, x \cdot \operatorname{ch} y$, при условии $f(0) = 3i$.

2) мнимой части $v(x,y) = 3x + 2xy$, при условии $f(-i) = 2$.

3) вещественной части $u(x,y) = x\cos\, x \cdot \operatorname{ch} y + y\sin\, x \cdot \operatorname{sh} y$, при условии $f(\pi) = -\pi$.

4) мнимой части $v(x,y) = 3e^{-y}\sin\, x + 1$, при условии $f(0) = 3 + i$.

5) вещественной части $u(x,y) = 2x^3 - 6xy^2 + 3x$, при условии $f(i) = 0$.

6) мнимой части $v(x,y) = 3\operatorname{sh} x \cdot \cos\, y + 1$, при условии $f(0) = i$.

7) вещественной части $u(x,y) = -3\operatorname{ch} x \cdot \sin\, y$, при условии $f(0) = i$.

8) мнимой части $v(x,y) = 6x^2y - 2y^3 + 3y - 1$, при условии $f(i) = 0$.

9) вещественной части $u(x,y) = x\sin\, x\operatorname{ch} y - y\cos\, x\operatorname{sh} y$, при условии $f(i) = -\operatorname{sh}1$.

10) мнимой части $v(x,y) = y - \sin\, x \cdot \operatorname{sh} y$, при условии $f(\pi) = \pi - 1$.

11) вещественной части $u(x,y) = x^3 - 3xy^2 + 3x^2 - 3y^2 + x$, если $f(0) = 0$.

12) мнимой части $v(x,y) = y\sin x \cdot \operatorname{ch} y + x\cos x \cdot \operatorname{sh} y$, если $f(-i) = -\operatorname{sh}1$.

13) вещественной части $u(x,y) = x + \cos x \cdot \operatorname{ch} y$, при условии $f(\pi) = \pi - 1$.

14) мнимой части $v(x,y) = 2\cos\, x \cdot \operatorname{sh} y + 3$, при условии $f(0) = 3i$.

15) вещественной части $u(x,y) = y\cos\, x\operatorname{ch} y - x\sin\, x\operatorname{sh} y$, при условии $f(\pi) = -\pi$.

16) мнимой части $v(x,y) = 3x^2y - y^3 + 6xy + y - 1$, если $f(-i) = 0$.

5 Особые точки функций комплексной переменной

5.1 Нули аналитических функций

Пусть функция $f(z)$ является аналитической в точке z_0.

Точка z_0 называется ***нулем функции*** $f(z)$, *если ее значение в этой точке равно нулю, т.е.*$f(z) = 0$.

В разложении функции в ряд Тейлора в окрестности нуля этой функции отсутствует свободный член: $c_0 = f(z_0) = 0$.

Если при этом в разложении отсутствуют и слагаемые, содержащие степени разности $(z - z_0)$ до k-й степени, т.е. разложение имеет вид

$$f(z) = \sum_{n=k}^{\infty} c_n (z - z_0)^n,\ c_n \neq 0,$$

то точка z_0 ***называется нулем кратности*** k *функции* $f(z)$.

Или, что эквивалентно предыдущему:

Нуль z_0 функции $f(z)$ называется ***нулем кратности*** k, *если*

$$f(z) = (z - z_0)^k g(z), \qquad g(z_0) \neq 0.$$

Теорема. *Число z_0 является нулем кратности k аналтитческой функции $f(z)$,*

$$\Leftrightarrow \qquad f(z_0) = f'(z_0) = f''(z_0) = \ldots = f^{(k-1)}(z_0) = 0, \qquad f^{(k)}(z_0) \neq 0.$$

Пример 13. *Найти все нули и их кратности функции* $f(z) = z\sinh(z)$.
Определяем рассматриваемую функцию:

```
> f:=z->z*sinh(z): `f(z)`=f(z);
```

$$f(z) = z\sinh(z)$$

Находим ее корни. Способов это сделать в Maple очень много, здесь приведен лишь один из возможных.

```
> kk:=allvalues(RootOf(f(z),z)): n:=nops(kk):
> print(`Множество корней f(z)`=subs(_Z1=k,kk),`где k любое целое
  число`);
```

Множество корней $f(z) = \{0,\ 2I\pi k, \pi I + 2I\pi k\}$, где k любое целое число

Для каждого из найденных корней, используя предыдущую теорему, находим кратность.

```
> for j from 1 to n do
> f1:=f(kk[j]):
> for i from 1 while (f1=0) do
> f1:=subs(z=kk[j],diff(f(z),z$i))
> end do:
> print(`Кратность корня z0`=subs(_Z1=k,kk[j]),` равна `,i-1);
> end do:
```

Кратность корня z0 $=0$, равна , 2

Кратность корня $z0=\pi I+2I\pi k$, равна , 1

Кратность корня $z0=2I\pi k$, равна , 1

Пример 14. *Найти порядок всех нулей функции* $\frac{(z^2+4)(z-2i)}{z^7}$ *(с учетом* ∞*).*

Первая часть этого примера аналогична предыдущему. Хотя, так как числителем $f(z)$ является многочлен, более рационально было бы использовать для нахождения корней функцию roots, которая для могочленов находит корни вместе с их кратностями.

```
> f:=z->((z^2+4)*(z-2*I))/(z^7): `f(z)`=f(z);
> k:={allvalues(RootOf(f(z),z))}; n:=nops(k):
> for j from 1 to n do
> f1:=f(k[j]):
> for i from 1 while (f1=0) do
> f1:=subs(z=k[j],diff(f(z),z$i))
> end do:print(`Кратность корня z0`=k[j],` равна`,(i-1));
> end do:
```

$$f(z)=\frac{(z^2+4)(z-2I)}{z^7}$$

$$k:=\{2I,-2I\}$$

Кратность корня z0= 2 I, равна, 2

Кратность корня z0=-2 I, равна, 1

Далее исследуем поведение функции $f(z)$ в бесконечно удаленной точке $z0=\infty$. Для этого удобно предварительно сделать замену $z=\frac{1}{t}$ и далее рассматривать функцию в точке $t=0$.

```
> l:=limit(factor(f(z)),z=infinity,complex):
> print('limit((f(z)),z=infinity)'=l);
> if l=0 then
> g(t):=factor((subs(z=1/t,f(z)))):
> f1:=subs(t=0,g(t)):
> for j from 1 while (f1=0) do
```

```
> f1:=subs(t=0,diff(g(t),t$j))
> end do:
> print(`Точка z0`=infinity,`корень функции f(z) кратности`,(j-1));
> else
> print(`Точка z0`=infinity,` не является корнем функции f(z)`);
> end if:
```

$$\lim_{z \to \infty} f(z) = 0$$

$$\text{Точка } z0 = \infty, \text{ корень функции } f(z) \text{ кратности } , 4$$

5.2 Нахождение особых точек и определение их типа у рациональных функций

Пусть функция $f(z)$ аналитическая в кольце $0 < |z - a| < \rho$, но не аналитическая в точке a ($a \neq \infty$). Тогда точка a называется *изолированной особой точкой однозначного характера для функции* $f(z)$.

В зависимости от поведения функции $f(z)$ вблизи точки a различают три типа особых точек.

Изолированная особая точка $a \in \overline{\mathbb{C}}$ однозначного характера функции $f(z)$ называется

- *устранимой особенностью,* если $\exists \lim\limits_{z \to a} f(z) \in \mathbb{C}$.
- *полюсом,* если $\exists \lim\limits_{z \to a} f(z) = \infty$.
- *существенной особенностью,* если $\nexists \lim\limits_{z \to a} f(z)$.

При исследовании функции в бесконечно удаленной точке $z0 = \infty$ удобно предварительно делать замену $z = \frac{1}{t}$ и далее рассматривать поведение функции в точке $t = 0$.

Для того чтобы точка $z_0 \in \overline{\mathbb{C}}$ была полюсом порядка n функции $f(z)$, необходимо и достаточно, чтобы она была нулем порядка n функции $1/f(z)$ (*связь нулей с полюсами*).

Пример 15. *Найти особые точки функции $f(z) = \frac{(z^2+1)^3}{(z+1)^3(z-i)}$, выяснить их характер (включая ∞).*

Заданная функция рациональная, следовательно у нее особенности могут быть лишь устранимые или полюса. Функция Maple limit работает корректно только в случае существования конечного предела, поэтому классифицируем по принципу получился конечный предел, значит устранимая особенность, если получилось что-нибудь другое (о том как работает limit и что может получится см., например, [1, с. 230]), значит – полюс.

Итак, задаем исходную функцию и упрощаем ее:

```
> f:=z->(z^2+1)^3/((z-I)*(z+1)^3):  'f(z)'=f(z);
```

$$f(z) = \frac{(z^2+1)^3}{(z-I)(z+1)^3}$$

```
> 'f(z)'=factor(f(z));
```

$$f(z) = \frac{(z+I)^3(z-I)^2}{(z+1)^3}$$

С помощью функции singular находим особенности функции $f(z)$ и nops считает их колличество. Обратите внимание: здесь формируется множество особенностей, хотя можно было бы вывести их список.

```
> s:=singular(f(z)); n:=nops(s);
```

$$s := \{\{z = I\}, \{z = -1\}, \{z = \infty\}, \{z = -\infty\}\}$$

$$n := 4$$

Далее для каждой особой точки $z0$ находим и печатаем (функция print) предел функции $f(z)$, не забывая указать, что нас интересует комплексный предел (опция complex). Если предел конечный печатаем, что точка $z0$ является устранимой особенностью, иначе – находим порядок полюса в точке $z0$, дифференцированием функции $1/f(z)$, и печатаем соответствующую информацию.

```
> for i from 1 to n do
> z0:=op(op(s[i]))[2]:
> if z0<>infinity and z0<>-infinity then
> l:=limit(factor(f(z)),z=z0,complex):
> print('limit((f(z)),z=z0)'=l);
> if -infinity<Re(l) and infinity>Re(l) then
> print(`В точке z0`=z0,`у функции f(z) устранимая особенность `);
> else
> g(z):=factor(1/f(z)):
> f1:=subs(z=z0,g(z)):
> for j from 1 while (f1=0) do
> f1:=subs(z=z0,diff(g(z),z$j))
> end do;
> print(`В точке z0`=z0,`у функции f(z) полюс, порядока `=(j-1));
> end if; end if;end do:
```

$$\lim_{z \to z0} f(z) = 0$$

$$\text{В точке } z0 = I, \text{ у функции } f(z) \text{ устранимая особенность}$$

$$\lim_{z \to z0} f(z) = \text{ undefined}$$

$$\text{В точке } z0 = -1, \text{ у функции } f(z) \text{ полюс, порядока} = 3$$

Все тоже самое проделываем для бесконечно удаленной точки.

```
> l:=limit(factor(f(z)), z=infinity, complex):
> print('limit((f(z)), z=infinity)'=l);
```

Выполним замену $z = \frac{1}{t}$ и будем исследовать поведение при $t = 0$.

```
> if -infinity<Re(l) and infinity>Re(l)
> then
> print(`в точке z0`=infinity,` у функции f(z) устранимая
  особенность`);
> else
> g(t):=factor(1/(subs(z=1/t,f(z))));
> f1:=subs(t=0,g(t)): j:=1:
> for j while (f1=0) do
> f1:=subs(t=0,diff(g(t),t$j))
> end do:
> print(`в точке z0`=z0,` у функции f(z) полюс, порядока `=(j-1));
> end if:
```

$$\lim_{z \to z0} f(z) = \infty + \infty I$$

$$\text{в точке } z0 = \infty, \text{ у функции } f(z) \text{ полюс, порядока} = 2$$

5.3 Ряды Лорана

Теорема Лорана. *Каждая функция $f(z)$, однозначная и аналитическая в круговом кольце $D : r < |z - z_0| < R$, представляется в этом кольце сходящимся рядом Лорана:*

$$f(z) = \sum_{n=-\infty}^{\infty} c_n (z - z_0)^n.$$

Пример 16. *Рассмотрим различные разложения в ряд Лорана функции*

$$f(z) = \frac{z+2}{z^2 - 2z - 3},$$

выбрав $z_0 = 0$.

Выбор 0 в качестве z_0, означает, что нужно получить разложение этой функции по степеням z. Функция $f(z)$ имеет две особые точки: $z = -1$ и $z = 3$. Следовательно, имеется три круговых «кольца» с центром в точке $z_0 = 0$, в каждом из которых функция аналитична, а именно:

1) круг $|z| < 1$,

2) кольцо $1 < |z| < 3$,

3) внешность круга $|z| > 3$.

Функцию $f(z)$ представим в виде суммы двух элементарных дробей:

```
> with(plots): with(numapprox):
> f:=z->(z+2)/(z^2-2*z-3): `f(z)`=f(z);
> `f(z)`= factor(f(z));
> w:=convert(factor(f(z)),parfrac,z):
> `f(z)`=w;
```

$$f(z) = \frac{z+2}{z^2 - 2z - 3}$$

$$f(z) = \frac{z+2}{(z+1)(z-3)}$$

$$f(z) = \frac{1}{4(z+1)} + \frac{5}{4(z-3)}$$

1) Разложение в круге $|z| < 1$.

Изобразим эту область, обозначив кружочками особые точки $z = -1$ и $z = 3$.

```
> z:=x+I*y:
> p1:=implicitplot(evalc(abs(z))<1,x=-4..4,y=-4..4,filled=true,
  coloring=[white,grey],linestyle=3,thickness=1):
> p2:=pointplot([-1,0],symbolsize=24,symbol=CIRCLE, color=black):
> p3:=pointplot([3,0],symbolsize=24,symbol=CIRCLE, color=black):
> display(p1,p2,p3);
```



Разложим функцию $f(z)$ в ряд Лорана с помощью функции laurent пакета numapprox, выбрав центром разложения 0 и указав, что нас интересуют первые 5 слагаемых.

```
> z:='z':
```

```
> `f(z)`=laurent(f(z),z=0,5);
```

$$f(z) = -\frac{2}{3} + \frac{1}{9}z - \frac{8}{27}z^2 + \frac{19}{81}z^3 - \frac{62}{243}z^4 + O(z^5)$$

В этом разложении отсутствуют отрицательные степени z, поэтому полученное разложение является рядом Тейлора.

2) Разложение в кольце $1 < |z| < 3$.

```
> z:=x+I*y:
> p1:=implicitplot(evalc(abs(z))<3,x=-4..4,y=-4..4,filled=true,
  coloring=[white,grey],linestyle=3,thickness=1):
> p2:=plot([sqrt(1-x^2),-sqrt(1-x^2)], x=-2..2,filled=true,
  color=[grey,grey]):
> p3:=plot([sqrt(1-x^2),-sqrt(1-x^2)], x=-2..2,y=-2..2,
  color=black,linestyle=3,thickness=1):
> p4:=pointplot([-1,0],symbolsize=24,symbol=CIRCLE, color=black):
> p5:=pointplot([3,0],symbolsize=24,symbol=CIRCLE,color=black):
> display(p2,p3,p1,p4,p5);
```



Здесь нужно раскладывать в ряд каждую из элементарных дробей в отдельности. Причем вторую будем раскладывать на бесконечности, так как ее ряд в нуле теперь уже будет расходящимся, а первую, как и прежде, в нуле.

```
> z:='z':
> op(2,w)=asympt(op(2,w),z,5);
```

$$-\frac{1}{4(z+1)} = -\frac{1}{4z} + \frac{1}{4z^2} - \frac{1}{4z^3} + \frac{1}{4z^4} + O\left(\frac{1}{z^5}\right)$$

```
> op(1,w)=laurent(op(1,w),z=0,5);
```

$$\frac{5}{4(z-3)} = -\frac{5}{12} - \frac{5}{36}z - \frac{5}{108}z^2 - \frac{5}{324}z^3 - \frac{5}{972}z^4 + O(z^5)$$

```
> `f(z)`=rhs(%+%%);
```

$f(z)=\left(-\frac{5}{12}-\frac{5}{36}z-\frac{5}{108}z^2-\frac{5}{324}z^3-\frac{5}{972}z^4+O(z^5)\right)-\frac{1}{4z}+\frac{1}{4z^2}-\frac{1}{4z^3}+\frac{1}{4z^4}+O(\frac{1}{z^5})$

3) Разложение в области $|z|>3$.

```
> z:=x+I*y:
> p1:=implicitplot(evalc(abs(z))>3,x=-4..4,y=-4..4,
  filled=true,coloring=[white,grey],linestyle=4,thickness=1):
> p2:=pointplot([3,0],symbolsize=24,symbol=CIRCLE,color=black):
> p3:=pointplot([-1,0],symbolsize=24,symbol=CIRCLE,color=black):
> display(p1,p2,p3);
```



Теперь нас интересует разложение функции $f(z)$ в ряд Лорана на бесконечности. Оно получается с помощью функции asympt.

```
> z:='z':
> `f(z)`=asympt(f(z),z,5);
```

$$f(z)=\frac{1}{z}+\frac{4}{z^2}+\frac{11}{z^3}+\frac{34}{z^4}+O\left(\frac{1}{z^5}\right)$$

Пример 17. *Записать разложение функции*

$$f(z)=\frac{z+2}{z^2-2z-3},$$

в окрестности точки -1.

Записать разложение в окрестности точки -1, означает, что нужно получить разложение этой функции по степеням $z+1$, т.е. $z_0=-1$. Расстояние до ближайшей особой точки $z=3$ равно четырем, поэтому окрестность точки $z_0=-1$ — это проколотая окрестность $0<|z+1|<4$.

```
> with(numapprox): with(plots):
> z0:=-1;
> f:=z->(z+2)/(z^2-2*z-3):  `f(z)`=f(z);
```

$$z0:=-1$$

$$f(z) = \frac{z+2}{z^2-2z-3}$$

```
> p1 := circle([-1,0], 4,color=black,thickness=2):
> p2 := circle([-1,0], 1/5,color=black):
> p3 := circle([3,0], 1/5,color=black):
> plots[display](p1,p2,p3,tickmarks=[10,10]);
```




```
> `f(z)`=laurent(f(z),z=z0,4);
```

$$f(z) = -\frac{1}{4}(z+1)^{-1} - \frac{5}{16} - \frac{5}{64}(z+1) - \frac{5}{256}(z+1)^2 - \frac{5}{1024}(z+1)^3 + O((z+1)^4)$$

5.4 Задачи для самостоятельного решения

1. Найти все нули функции $f(z)$ и определить их кратности.

1)	$f(z) = (z^2+\pi^2)(e^{-z}-1)$,	2)	$\sin\ z + \operatorname{sh}\ iz$.
3)	$f(z) = z^2 \sin\ z$,	4)	$z \operatorname{sh}\ iz$.
5)	$f(z) = 1 + \operatorname{ch}\ z$,	6)	$(1-\operatorname{sh}\ z)^2$.
7)	$f(z) = (z+\pi i)\operatorname{sh}\ z$,	8)	$\cos\ z^3$.
9)	$f(z) = (z^2+\pi^2)\sin\ z$,	10)	$\cos\ z + \operatorname{ch}\ iz$.
11)	$f(z) = z\sin^2\ z$,	12)	$z\operatorname{sh}^2\ z$.
13)	$f(z) = (z+\pi)\operatorname{sh}\ iz$,	14)	$z^3(\cos\ z - 1)$.
15)	$f(z) = (z^2+\pi^2)(e^z+1)$,	16)	$\sin\ z + i\operatorname{sh}\ iz$.

2. Найти порядок всех нулей функции $f(z)$ (с учетом ∞).

1) $f(z)=\frac{(z^3-1)(z-1)z}{(z-1)^7-3}$,	2) $f(z)=\frac{(z^2-4)z}{(z-2)^9-3}$,
3) $f(z)=\frac{(z^2-1)(4z^2+1)z^2}{(z-5)^2((z-1)^7-3)}$,	4) $f(z)=\frac{(z^2+1)(z^2-1)}{(z-3)^9-2}$,
5) $f(z)=\frac{(z^2+4)z(z^5-i)}{(z+1)^3(z^2-4)}$,	6) $f(z)=\frac{(z^6+3)(z-4)}{(z^2-4)(z^6+1)}$,
7) $f(z)=\frac{((z-1)^7-3)(z-2)}{(z^2-4)z(z^6-i)}$,	8) $f(z)=\frac{(z^2-1)(4z^2+1)z^2}{(z-1)^2(z^2+1)}$,
9) $f(z)=\frac{(z-2)((z-3)^9-2)}{(z^2-4)z}$,	10) $f(z)=\frac{(z^6+64)(z^2+4)}{(z^2+4)(z^2-4)}$,
11) $f(z)=\frac{(2z-1)^3(z+1)^2}{(z^2-1)(4z^2+1)z^2}$,	12) $f(z)=\frac{(z^2-4)(z^6+1)}{(z^2+1)(z^2-1)}$,
13) $f(z)=\frac{(z^2+4)(z^2-4)}{(z^2-1)z^2(z^2+1)(z-10)}$,	14) $f(z)=\frac{(z+1)^3(z^2-4)}{(z^2+4)z(z^5-i)}$,
15) $f(z)=\frac{(z-5)^2((z-1)^7-3)}{(z^3-1)(z-5i)}$,	16) $f(z)=\frac{z^2+2iz}{(z^6+3)(z^2+4)}$.

3. Найти особые точки функции $f(z)$ (см. предыдущую задачу) выяснить их характер (включая ∞).

4. Разложить функцию $f(z)$ в ряд Лорана в кольце $R<|z|<R_1$ и в окрестности точки z_1.

1) $f(z)=\frac{z^2}{(z-1)(z^2-4)}$,	$R=1$,	$R_1=2$,	$z_1=\infty$.
2) $f(z)=\frac{(z+1)^2}{(z^2-1)(4z^2+1)}$,	$R=1/2$,	$R_1=1$,	$z_1=i/2$.
3) $f(z)=\frac{z}{z^3-z^2+5+3z}$,	$R=1$,	$R_1=\sqrt{5}$,	$z_1=2$.
4) $f(z)=\frac{1}{z^3-z^2+4z-4}$,	$R=1$,	$R_1=2$,	$z_1=2$.
5) $f(z)=\frac{1}{(z-1)^2(z^2+1)}$,	$R=0$,	$R_1=1$,	$z_1=\infty$.
6) $f(z)=\frac{z+3}{(z+1)(z^2-4)}$,	$R=1$,	$R_1=2$,	$z_1=2$.
7) $f(z)=\frac{z+4}{z^2-z-6}$,	$R=2$,	$R_1=3$,	$z_1=3$.
8) $f(z)=\frac{z}{(z^2+4)(z-1)}$,	$R=1$,	$R_1=2$,	$z_1=-2$.
9) $f(z)=\frac{1}{(z^2+4)(z-3i)}$,	$R=2$,	$R_1=3$,	$z_1=3i$.
10) $f(z)=\frac{1}{(z-2)^2(z^2+1)}$,	$R=1$,	$R_1=2$,	$z_1=i$.
11) $f(z)=\frac{1}{(z-1)(z^2+z-2)}$,	$R=1$,	$R_1=2$,	$z_1=\infty$.
12) $f(z)=\frac{z-4}{(z^2+4)(z+1)}$,	$R=1,;$	$R_1=2$,	$z_1=-2$.
13) $f(z)=\frac{z+1}{(z^2-9)z(z-8)}$,	$R=3$,	$R_1=8$,	$z_1=-3$.
14) $f(z)=\frac{1}{(z-2)(z^2+1)}$,	$R=1$,	$R_1=2$,	$z_1=i$.
15) $f(z)=\frac{z+1}{z^3+z^2+4z+4}$,	$R=1$,	$R_1=2$,	$z_1=2i$.
16) $f(z)=\frac{z+10}{(z^2-4)(z-3)}$,	$R=2$,	$R_1=3$,	$z_1=-2$.

6 Интегрирование

6.1 Вычисление интегралов с помощью вычетов и разложения в ряд Лорана

Для того чтобы вычислить контурный интеграл от функции комплексной переменной аналитически нужно:

1) найти все ее особые точки;

2) выбрать, среди них те, которые попали внутрь контура интегрирования;

3) определить тип особенности в каждой точке;

4) в зависимости от типа, вычислить вычеты (для каждого типа особенности своя формула для подсчета вычета, см.,например, [11, занятие 7]);

5) и, наконец, найти значение интеграла, воспользовавшись теоремой Коши о вычетах.

Теорема Коши о вычетах:
$f(z)$ аналитическая в области D и на ее границе ∂D, кроме конечного числа точек $\{z_1, \dots z_n\}$, лежащих внутри D,
$\Longrightarrow$

$$\oint\limits_{\partial D} f(z)dz = 2\pi i \sum_{k=1}^{n} \underset{z_k}{res}\, f(z).$$

Хотя, в некоторых случаях, проще получить результат, используя особые точки, не попавшие внутрь контура интегрирования (см. пример 19).

Теорема 1.
$f(z)$ аналитическая в $\mathbb{C}$ кроме конечного числа точек $\{z_1, \dots z_n\} \subset \mathbb{C}$
$\Longrightarrow$

$$\sum_{k=1}^{n} \underset{z_k}{res}\, f(z) + \underset{\infty}{res}\, f(z) = 0.$$

Используя Maple, особенности будем находить с помощью singular, их количество — nops, вычеты — residue. Таким образом, выяснять тип особенности и знать формулы им соответствующие для вычетов не нужно. При этом, в существенно особой точке все же придется поступить иначе (см. пример 21).

Пример 18. *Вычислить интеграл* $\oint\limits_{|z+2i|=3} \frac{dz}{z^3(z^2+4)^2}$, *считая, что обход контура происходит в положительном направлении.*

Обозначим zc–центр окружности интегрирования, R– ее радиус, C– окружность интегрирования:

```
> with(plots): with(plottools):
> f:=z->1/((z^3)*(z^2+4)^2): 'f(z)'=f(z);
> zc:=-2*I: R:=3: C:=abs(z-zc)=R:
```

$$f(z) = \frac{1}{z^3(z^2+4)^2}$$

Вспомним и выведем на экран теорему Коши о вычетах (суммирование идет по всем изилированным особым точкам попавшим внутрь контура интегрирования). Обратите внимание, что в Maple принято обозначать пределы интегрирования несколько справа от интеграла, а не четко над и под ним.

```
> `Теорема Коши о вычетах:`;
> Int('f(z)',z='C'..``)=2*Pi*I*Sum(res[z=z[k]]('f(z)'),k);
> Int(f(z),z=C..``)=2*Pi*I*Sum(res[z=z[k]](f(z)),k);
```

Теорема Коши о вычетах:

$$\int_C f(z)dz = 2I\pi \left(\sum_k \mathrm{res}_{z=z_k}\left(f(z)\right) \right)$$

$$\int_{|z+2I|=3} \frac{1}{z^3(z^2+4)^2} dz = 2I\pi \left(\sum_k \mathrm{res}_{z=z_k}\left(\frac{1}{z^3(z^2+4)^2} \right) \right)$$

Найдем особые точки данной функции и их количество. Функция `nops` умеет находить количество элементов множества, или списка, в данном случае удобнее список.

```
> zzz:=[singular(f(z))];n:=nops(zzz);
```

$$zzz := [\{z=0\}, \{z=2I\}, \{z=-2I\}]$$

$$n := 3$$

```
> `Особые точки функции f(z):`
> for j from 1 to n do
> z[j]:=op(zzz[j][1])[2];
> end do;
```

Особые точки функции f(z):

$$z_1 := 0$$
$$z_2 := 2I$$
$$z_3 := -2I$$

Рисуем контур интегрирования и особые точки.

```
> for j from 1 to n do
> pp||j:=textplot([(Re(z[j]))+0.4,Im(z[j])-0.1,"z"||j],align=BELOW):
```

```
> pp||(n+j):=circle([Re(z[j]),Im(z[j])],0.05,color=black,thickness=3):
> end do:
> pp0:=implicitplot(abs(x+I*y-zc)=R,x=-5..5,y=-5..5,color=black):
> display(seq(pp||i,i=0..2*n));
```



Суммируем вычеты в тех особых точках, которые попали внутрь контура интегрирования.

```
> zinC:=0: S:=0:
> for j from 1 to n do
> if evalf(evalc(abs(z[j]-zc)))<R then
> zinC:=zinC+1:
> S:=S+residue(f(z),z=z[j]);
> end if;
> end do;
> `Количество особых точек попавших внутрь контура интегрирования`
   = zinC;
> Int('f(z)',z=C..``)=2*Pi*I*S;
```

Количество особых точек попавших внутрь контура интегрирования $= 2$

$$\int_{|z+2I|=3} f(z)dz = \frac{-1}{32} I\pi$$

Пример 19. *Вычислить интеграл* $\oint\limits_{|z|=2} \frac{dz}{(z-3)(z^4+2)}$, *считая, что обход контура происходит в положительном направлении.*

```
> with(plots): with(plottools):
> f:=z->1/((z-3)*(z^4+2)): zc:=0: R:=2:
> `f(z)`=f(z); C:=abs(z-zc)=R:
```

$$f(z) = \frac{1}{(z-3)(z^4+2)}$$

```
> `Теорема Коши о вычетах.`
> Int('f(z)',z='C'..``)=2*Pi*I*Sum(res[z=z[k]]('f(z)'),k);
> Int(f(z),z=C..``)=2*Pi*I*Sum(res[z=z[k]](f(z)),k);
```

Теорема Коши о вычетах.

$$\int_C f(z)dz = 2I\pi\left(\sum_k \mathrm{res}_{z=z_k}\left(f(z)\right)\right)$$

$$\int_{|z|=2} \frac{1}{(z-3)(z^4+2)}dz = 2I\pi\left(\sum_k \mathrm{res}_{z=z_k}\left(\frac{1}{(z-3)(z^4+2)}\right)\right)$$

```
> `Особые точки функции f(z):`;
> zzz:=[evalf(singular(f(z)))]: n:=nops(zzz):
> for j from 1 to n do
> z[j]:=op(zzz[j][1])[2]; end do;
```

Особые точки функции f(z):

$$z_1 := 3$$

$$z_2 := \frac{2^{\frac{3}{4}}}{2} + \frac{1}{2}I2^{\frac{3}{4}}$$

$$z_3 := -\frac{2^{\frac{3}{4}}}{2} + \frac{1}{2}I2^{\frac{3}{4}}$$

$$z_4 := -\frac{2^{\frac{3}{4}}}{2} - \frac{1}{2}I2^{\frac{3}{4}}$$

$$z_5 := \frac{2^{\frac{3}{4}}}{2} - \frac{1}{2}I2^{\frac{3}{4}}$$

```
> pp0:=implicitplot(abs(x+I*y-zc)=R, x=-5..5,y=-5..5, color=black,
  thickness=2):
> for j from 1 to n do
> pp||j:=textplot([Re(z[j]),Im(z[j])+0.2,"z"||j],align=ABOVE):
> pp||(n+j):=circle([Re(z[j]),Im(z[j])], 0.03, color=black,
  thickness=3):
> end do:
> p1:=pointplot([Re(zc),Im(zc)],color=black,symbol=CROSS):
> p2:=textplot([Re(zc)+0.2,Im(zc)+0.1,"zc"],align=ABOVE):
> display(seq(pp||i,i=0..2*n),p1,p2);
```



```
> zinC:=0: S:=0: S1:=0:
> for j from 1 to n do
> if evalf[5](evalc(abs(z[j]-zc)))<R then
> zinC:=zinC+1:
> S:=S+residue(f(z),z=z[j]);
> else
> S1:=S1+residue(f(z),z=z[j]):
> end if;
> end do;
> `Количество особых точек попавших внутрь контура интегрирования`
   = zinC;
> Int('f(z)',z='C'..``)=2*Pi*I*S;
> `Количество особых точек не попавших внутрь контура интегрирования`
  = n-zinC;
> Int('f(z)',z='C'..``)= -2*Pi*I*(S1+residue(f(z),z=infinity));
```

Количество особых точек попавших внутрь контура интегрирования $= 4$

$$\int_C f(z)dz = (-0. - .2409638552I)\pi$$

Количество особых точек не попавших внутрь контура интегрирования $= 1$

$$\int_C f(z)dz = \frac{-2}{83} I\pi$$

```
> `Проверка:`;
> Sum(res[z=z[k]]('f(z)'),k)+'res[z=infinity](f(z))'=
  =simplify(evalc(S+S1+residue(f(z),z=infinity)));
```

Проверка:

$$\left(\sum_k \mathrm{res}_{z=z_k}\left(f(z)\right)\right) + \mathrm{res}_{z=\infty}(f(z)) = 0$$

Этот пример выполнен аналогично предыдущему примеру 18, хотя это не рационально в данном случае. Здесь 4 особые точки попали внутрь контура интегрирования и 2, учитывая бесконечно удаленную, находятся вне его. Поэтому, можно найти интеграл, вычислив только лишь вычеты в точке $z_1 = 3$ и в бесконечно удаленной.

Такой подход ничего бы не упростил бы в предыдущем примере 18, так как там 2 особые точки внутри контура интегрирования и 2 вне его, включая бесконечно удаленную точку.

Далее приведем пример, который выполнить используя особые точки внутри контура не удается.

Пример 20. *Вычислить интеграл* $\oint\limits_{|z|=2} \frac{dz}{z^{10}-2}$, *считая, что обход контура происходит в положительном направлении.*

```
> with(plots): with(plottools):
> f:=z->1/(z^10-2): zc:=0: R:=2:
> 'f(z)'=f(z); C:=abs(z-zc)=R:
```

$$f(z) = \frac{1}{z^{10}-2}$$

```
> Int('f(z)',z='C'..``)=2*Pi*I*Sum(res[z=z[k]]('f(z)'),k);
> Int(f(z),z=C..``)=2*Pi*I*Sum(res[z=z[k]](f(z)),k);
```

$$\int_C f(z)dz = 2I\pi\left(\sum_k \mathrm{res}_{z=z_k}\left(f(z)\right)\right)$$

$$\int_{|z|=2} \frac{1}{z^{10}-2}dz = 2I\pi\left(\sum_k \mathrm{res}_{z=z_k}\left(\frac{1}{z^{10}-2}\right)\right)$$

Найдем особые точки данной функции и их количество.

```
> zzz:=[evalf(singular(f(z)))]: n:=nops(zzz):
> for j from 1 to n do
> z[j]:=op(zzz[j][1])[2]: end do:
```

Рисуем контур интегрирования и особые точки.

```
> pp0:=implicitplot(abs(x+I*y-zc)=R, x=-5..5,y=-5..5, color=black,
  thickness=2):
> for j from 1 to n do
> pp||j:=textplot([Re(z[j]),Im(z[j])+0.2,"z"||j],align=ABOVE):
> pp||(n+j):=circle([Re(z[j]),Im(z[j])], 0.03, color=black,
  thickness=3): end do:
> p1:=pointplot([Re(zc),Im(zc)],color=black,symbol=CROSS):
> p2:=textplot([Re(zc)+0.2,Im(zc)+0.1,"zc"],align=ABOVE):
> display(seq(pp||i,i=0..2*n),p1,p2);
```



Здесь 10 особых точек и они все попали внутрь контура интегрирования. Поэтому, воспользовавшись теоремой 1 (стр. 49), можно найти интеграл, вычислив только лишь вычет в бесконечно удаленной точке.

```
> Int(f(z),z=C..``)=-2*Pi*I*residue(f(z),z=infinity);
```

$$\int_{|z|=2} \frac{1}{z^{10}-2} dz = 0$$

Пример 21. *Вычислить интеграл* $\oint\limits_{|z|=1} z^2 e^{1/z} dz$, *считая, что обход контура происходит в положительном направлении.*

Если попытаться воспользоваться предыдущим алгоритмом и в этом случае, возникнут несколько сложностей. Как легко видеть, у данной функции особыми точками являются ноль и бесконечно удаленная точка. Внутрь контура интегрирования попал ноль, в нем существенная особенность (см., например, [11, примеры 7.3. и 7.5.]). Поэтому вычет будем находить как c_{-1} коэффициент в разложении в ряд Лорана функции в окрестности нуля. Разложение в ряд Лорана получено с помощью laurent.

```
> with(numapprox):
> f:=z->z^2*exp(1/z): `f(z)`=f(z);
```

$$f(z) = z^2 e^{\frac{1}{z}}$$

Вспомним и напечатаем формулу необходимую для подсчета этого интеграла.

```
> Int(f(z),z=(abs(z)=1)..``)=2*Pi*I*c[-1];
```

$$\int_{|z|=1} z^2 e^{\frac{1}{z}} dz = 2I\pi c_{-1}$$

Выполняем замену переменной, подставив $z = \frac{1}{t}$. Ракладываем в ряд Лорана полученную функцию и делаем обратную замену переменной.

```
> `f(t)`=subs(z=1/t,f(z));
> `f(t)`=expand(laurent(rhs(%),t=0,6));
> `f(z)`=subs(t=1/z,rhs(%));
```

$$f(t) = \frac{e^t}{t^2}$$

$$f(t) = t^{-2} + t^{-1} + \frac{1}{2} + \frac{1}{6}t + \frac{1}{24}t^2 + \frac{1}{120}t^3 + O(t^4)$$

$$f(t) = z^2 + z + \frac{1}{2} + \frac{1}{6}\frac{1}{z} + \frac{1}{24}\frac{1}{z^2} + \frac{1}{120}\frac{1}{z^3} + O(\frac{1}{z^4})$$

Находим коэффициент c_{-1} равный вычету функции в существенно особой точке ноль.

```
> c[-1]:=coeff(convert(rhs(%),polynom),z^(-1));
```

$$c_{-1} := \frac{1}{6}$$

Получаем ответ:

```
> Int(f(z),z=(abs(z)=1)..``)=2*Pi*I*c[-1];
```

$$\int_{|z|=1} z^2 e^{\frac{1}{z}} dz = \frac{1}{3} I\pi$$

6.2 Задачи для самостоятельного решения.

Вычислить с помощью вычетов контурные интегралы

1) $\oint\limits_{|z-z_1|=R_1} f(z)dz$, **2)** $\oint\limits_{|z-z_2|=R_2} f(z)dz$, **3)** $\oint\limits_{|z-z_3|=R_3} g(z)dz$, **4)** $\oint\limits_{|z-z_4|=R_4} h(z)dz$.

1) $f(z) = \frac{z^3}{(z-3)(z^4-1)}$, $g(z) = \frac{z^3(z+1)^3}{(z-3)(z^4-1)}$, $h(z) = \frac{ze^{1/(3z)}}{z+3}$,
$R_1 = 1.2,\ z_1 = 1+i$; $R_3 = 2,\ z_3 = 0$; $R_4 = 4,\ z_4 = 0$.
$R_2 = 2,\ z_2 = 0$;

2) $f(z) = \frac{1-z^4+3z^6}{2z^5(z^4-1)}$, $g(z) = \frac{1-z^4+3z^6}{2z^5}$, $h(z) = \cos\left(\cos\frac{1}{z}\right)$,
$R_1 = 1.2,\ z_1 = 1+i$; $R_3 = 0.5,\ z_3 = 0$; $R_4 = \pi,\ z_4 = 0$.
$R_2 = 1.5,\ z_2 = i$;

3) $f(z) = \frac{z^2}{(z+1)^3(z^2-4)}$, $g(z) = \frac{z^2(z+2)^2}{(z+1)^3(z^2-4)}$, $h(z) = ze^{-1/z^2}$,
$R_1 = 1.2,\ z_1 = 1+i$; $R_3 = 2,\ z_3 = 0$; $R_4 = 1,\ z_4 = 0$.
$R_2 = 2,\ z_2 = 0$;

4) $f(z) = \frac{3+z^2+z^4}{2z^3(z^3-8)}$, $g(z) = \frac{3+z^2+z^4}{2z^3}$, $h(z) = z\cos\left(\frac{1}{z-i}\right)$,
$R_1 = 2,\ z_1 = 1+i$; $R_3 = 0.5,\ z_3 = 0$; $R_4 = \pi/4,\ z_4 = i$.
$R_2 = 2.5,\ z_2 = -1$;

5) $f(z) = \frac{z^2-3z}{(z-4)(z^4+1)}$, $g(z) = \frac{(z^2-3z)^6}{(z-4)(z^4+1)}$, $h(z) = \frac{e^{z/(z-i)}}{z-i}$,
$R_1 = 1,\ z_1 = 1$; $R_3 = 5,\ z_3 = 0$; $R_4 = \pi/4,\ z_4 = i$.
$R_2 = 2,\ z_2 = 0$;

6) $f(z) = \frac{3z^3-z^2+2}{4z^3(z^3+64)}$, $g(z) = \frac{3z^3-z^2+2}{4z^3}$, $h(z) = z^3 + \frac{1}{z} + \sin\frac{1}{z^2}$,
$R_1 = 2, z_1 = 1+i$; $R_3 = 0.5, z_3 = 0$; $R_4 = 1,\ \ z_4 = 0$.
$R_2 = 3.7,\ z_2 = 1+i$;

7) $f(z) = \frac{1}{(z-5i)(z^3-1)}$, $g(z) = \frac{z^5}{(z-5i)(z^3-1)}$, $h(z) = \sin\frac{z+1}{z-2}$,
$R_1 = 1, z_1 = 1$; $R_3 = 3, z_3 = 1+0.5i$; $R_4 = 1, z_4 = 2$.
$R_2 = 3, z_2 = 1+0.5i$;

8) $f(z) = \frac{1}{(z^2+4)z}$, $R_1 = 1.5, z_1 = i$; $R_2 = 2, z_2 = -i$; $g(z) = \frac{(z-2i)^4}{(z^2+4)z}$, $R_3 = 3, z_3 = 0$; $h(z) = \sin\left(\sin\frac{1}{z}\right)$, $R_4 = 1,\ z_4 = 0$.

9) $f(z) = \frac{1}{(z^2+i)(z^3+1)}$, $R_1 = 1,\ z_1 = 1$; $R_2 = 2.3,\ z_2 = -1+i$; $g(z) = \frac{z^5}{(z^2+i)(z^3+1)}$, $R_3 = 3,\ z_3 = 1+0.5i$; $h(z) = z\cos\frac{z-1}{z-i}$, $R_4 = 1,\ z_4 = i$.

10) $f(z) = \frac{5-2z^3+3z^4}{z^4(z^2+0.25)}$, $R_1 = 0.7,\ z_1 = 0.5i$; $R_2 = 2,\ z_2 = 1$; $g(z) = \frac{5-2z^3+3z^4}{z^4}$, $R_3 = 1,\ z_3 = 0$; $h(z) = \cos\frac{1}{z}$, $R_4 = 1,\ z_4 = 0$.

11) $f(z) = \frac{1}{z^3(z^2+4)(z^4-1)}$, $R_1 = 1.5,\ z_1 = 1+0.5i$; $R_2 = 3,\ z_2 = 1+i$; $g(z) = \frac{(z+1)^12}{z^3(z^2+4)(z^4-1)}$, $R_3 = 2,\ z_3 = 1+0.5i$; $h(z) = z\sin\frac{1}{z-3}$, $R_4 = 2,\ z_4 = 3$.

12) $f(z) = \frac{1+2z+3z^2+4z^3}{2z^2}$, $R_1 = 0.5,\ z_1 = 0$; $R_2 = 2,\ z_2 = 1$; $g(z) = \frac{(1+2z+3z^2+4z^3)^2}{2z^2}$, $R_3 = 0.5,\ z_3 = 0$; $h(z) = \sin^2\frac{1}{z}$, $R_4 = 1,\ z_4 = 0$.

13) $f(z) = \frac{2}{z^2(z-1)}$, $R_1 = 1.3,\ z_1 = 1+i$; $R_2 = 2,\ z_2 = 1$; $g(z) = \frac{2(z-i)^4}{z^2(z-1)}$, $R_3 = 1.5,\ z_3 = 0$; $h(z) = (z-3)e^{\frac{z}{(3-z)}}$, $R_4 = 4,\ z_4 = 3$.

14) $f(z) = \frac{1}{(z^3+1)(z^2-i)}$, $R_1 = 1.5,\ z_1 = 1$; $R_2 = 2,\ z_2 = -1+i$; $g(z) = \frac{z^7}{(z^3+1)(z^2-i)}$, $R_3 = 3,\ z_3 = 1+0.5i$; $h(z) = z\sin^2\frac{1}{z}$, $R_4 = 1,\ z_4 = 0$.

15) $f(z) = \frac{1}{z^2(z+1)^2(z^4-16)}$, $R_1 = 1.5,\ z_1 = 1+0.5i$; $R_2 = 2.5,\ z_2 = 1+i$; $g(z) = \frac{(z-2)^{12}}{z^2(z+1)^2(z^4-16)}$, $R_3 = 3.5,\ z_3 = 1+0.5i$; $h(z) = \cos^2\frac{1}{z}$, $R_4 = 1,\ z_4 = 0$.

16) $f(z) = \frac{1}{z(z^2+1)}$, $R_1 = 0.5,\ z_1 = 0$; $R_2 = 2,\ z_2 = 1$; $g(z) = \frac{(z-i)^4}{z(z^2+1)}$, $R_3 = 1.5,\ z_3 = 0$; $h(z) = z\cos^2\frac{1}{z}$, $R_4 = 1,\ z_4 = 0$.

7 Операционное исчисление. Преобразование Лапласа

7.1 Примеры

Пример 22. *Найти решения уравнения:* $\begin{cases} y'' - 3y' + 2y = e^t, \\ y(0) = 0,\ y'(0) = 0; \end{cases}$

Зададим условия задачи:

```
> odu:=diff(y(t),t$2)-3*diff(y(t),t)+2*y(t)=exp(t);
> ic:=y(0)=0, D(y)(0)=0;
```

$$odu := \left(\frac{d^2}{dt^2}\, y(t)\right) - 3\left(\frac{d}{dt}\, y(t)\right) + 2y(t) = e^t$$

$$ic := y(0) = 0,\ D(y)(0) = 0$$

1) Решим это дифференциальное уравнение, не указывая метод решения.

```
> dsolve({odu,ic},y(t)): expand(%);
```

$$y(t) = -e^t t + (e^t)^2 - e^t$$

Проверим полученное решение:

```
> assign(%%); simplify(odu);
```

$$e^t = e^t$$

2) Решим данное дифференциальное уравнение операционным методом.

```
>dsolve({odu,ic},y(t),method=laplace): expand(%);
```

$$y(t) = -e^t - te^t + (e^t)^2$$

Как и следовало ожидать, ответ получился такой же как и в предыдущем варианте решения.

3) При этом Maple позволяет построить график решения дифференциального уравнения без нахождения вида решения.

```
> with(DEtools): DEplot(odu,y(t),t=-2..2,y=-1..3,[[ic]]);
```



4) Посмотрим теперь подробное решение этого уравнения операционным методом.

```
> with(inttrans):
> odu:=diff(y(t),t$2)-3*diff(y(t),t)+2*y(t)=exp(t);
> y(0):=0;  D(y)(0):=0;
```

Функция `laplace` находит изображения сразу всех слагаемых входящих в данное дифференциальное уравнение, хотя вид получается несколько непривычный.

```
> laplace(odu,t,p);
```

$$odu := \left(\frac{d^2}{dt^2}\,y(t)\right) - 3\left(\frac{d}{dt}\,y(t)\right) + 2y(t) = e^t$$

$$y(0) := 0;$$

$$D(y)(0) := 0$$

$$p^2 \operatorname{laplace}(y(t), t, p) - 3p \operatorname{laplace}(y(t), t, p)1 + 2 \operatorname{laplace}(y(t), t, p) = \frac{1}{p-1}$$

Выполним подстановку в последнем уравнении, чтобы получить обычный вид уравнения в изображениях.

```
> subs(laplace(y(t),t,p)=Y(p),%);
```

$$p^2 Y(p) - 3pY(p) + 2Y(p) = \frac{1}{p-1}$$

Решаем полученное линейное уравнение

```
> Y(p):=solve(%,Y(p));
```

$$Y(p) := \frac{1}{p^3 - 4p^2 + 5p - 2}$$

Раскладываем на множители знаменатель полученной дроби

```
> Y(p):= factor(Y(p));
```

$$Y(p) := \frac{1}{(p-2)(p-1)^2}$$

Далее раскладываем $Y(p)$ в сумму простых дробей

```
> Y(p):=convert(Y(p),parfrac,p);
```

$$Y(p) := \frac{1}{p-2} - \frac{1}{p-1} - \frac{1}{(p-1)^2}$$

Делая обратное преобразование Лапласа, получаем окончательный ответ:

```
> y:=t->invlaplace(Y(p),p,t): y=y(t);
```

$$y = e^{(2t)} + (-1-t)e^t$$

Далее рассмотрим пример столь же подробного решения линейного дифференциального уравнения, но не с постоянными коэффициэнтами, а с переменными. Главным отличием будет то, что в изображениях получится не алгебраическое линейное уравнение, а линейное дифференциальное уравнение относительно изображения искомой функции.

Пример 23. *Найти решения уравнения:* $ty''(t)+(2t-1)y'(t)+(t-1)y(t)=0$.

Зададим условия задачи:

```
> odu:=t*diff(y(t),t$ 2)+(2*t-1)*diff(y(t),t)+(t-1)*y(t)=0;
```

$$odu := t\left(\frac{d^2}{dt^2}\,y(t)\right)+(2t-1)\left(\frac{d}{dt}\,y(t)\right)+(t-1)y(t)=0$$

1) Решим это дифференциальное уравнение, не указывая метод решения.

```
> dsolve(odu,y(t));
```

$$\{y(t) = \_C1\,e^{-t} + \_C2\,e^{-t}t^2\}$$

Проверим полученное решение:

```
> assign(%%); simplify(odu);
```

$$0=0$$

2) Решим данное дифференциальное уравнение операционным методом.

```
> dsolve(odu,y(t),method=laplace): expand(%);
```

$$y(t) = \left(-\frac{1}{2}(y(0)-\_C1)t^2 + y(0)\right)e^{-t}$$

Ответ в точности совпадает с полученным в предыдущем варианте решения, так как $-\frac{1}{2}(y(0)-\_C1)$ и $y(0)$ так же произвольные константы, лишь иначе обозначенные.

3) Посмотрим теперь подробное решение этого уравнения операционным методом.

```
> with(inttrans):
> odu:=t*diff(y(t),t$ 2)+(2*t-1)*diff(y(t),t)+(t-1)*y(t)=0;
> y(0):=0; D(y)(0):=0;
> laplace(odu,t,p);
```

$$odu := t\left(\frac{d^2}{dt^2}\,y(t)\right)+(2t-1)\left(\frac{d}{dt}\,y(t)\right)+(t-1)y(t)=0$$

$$-3p\,\mathrm{laplace}(y(t),t,p) - p^2\,\left(\frac{\partial}{\partial p}\mathrm{laplace}(y(t),t,p)\right)1 + 2y(0) -$$

$$-3\,\mathrm{laplace}(y(t),t,p) - 2p\left(\frac{\partial}{\partial p}\mathrm{laplace}(y(t),t,p)\right) - \left(\frac{\partial}{\partial p}\mathrm{laplace}(y(t),t,p)\right) = 0$$

Выполним преобразование Лапласа сразу всех слагаемых уравнения.

```
> subs(laplace(y(t),t,p)=Y(p),%);
```

$$-3pY(p) - p^2\left(\frac{d}{dp}Y(p)\right) + 2y(0) - 3Y(p) - 2p\left(\frac{d}{dp}Y(p)\right) - \left(\frac{d}{dp}Y(p)\right) = 0$$

Решаем полученное уравнение относительно $Y(p)$ с помощью функции `dsolve`, так как уравнение дифференциальное и раскадываем полученное решение в сумму простейших дробей.

```
> dsolve(%,Y(p));
> convert(%,parfrac,p); assign(%);
```

$$Y(p) = \frac{2y(0)\left(p + \frac{1}{2}p^2\right) + \_C1}{(1+p)^3}$$

$$Y(p) = \frac{y(0)}{1+p} + \frac{-y(0) + \_C1}{(1+p)^3}$$

Делая обратное преобразование Лапласа, получаем окончательный ответ:

```
> y:=t->invlaplace(Y(p),p,t): y=y(t);
```

$$y = \left(\frac{1}{2}(-y(0) + \_C1)t^2 + y(0)\right)e^{-t}$$

Пример 24. *Решить задачу Коши*

$$y'' + y = f(t), \qquad y(0) = y'(0) = 0,$$

если функция $f(t)$ задана графически



Пусть $\eta(t) = \begin{cases} 1, & t > 0 \\ 0, & t < 0 \end{cases}$ – функция Хевисайда, тогда

$$f(t) = \eta(t) - 2\eta(t-1) + \eta(t-2).$$

В Maple функция Хевисайда обозначают `Heaviside(t)`.

Зададим теперь условия задачи

```
> with(inttrans): with(plottools):
> odu:=diff(y(t),t$2)+y(t)=f(t);
> y(0):=0; D(y)(0):=0;
> f:=t->Heaviside(t)-2*Heaviside(t-1)+Heaviside(t-2);
```

$$odu := \left(\frac{d^2}{dt^2}y(t)\right) + y(t) = f(t)$$

$$y(0) := 0 \quad D(y)(0) := 0$$

$$f := t- > Heaviside(t) - 2Heaviside(t-1) + Heaviside(t-2)$$

Построим функцию $f(t)$ и увидим, что аналитическое выражение для нее было написано правильно. Хотя, конечно, никакой рисунок не является доказательством!

```
> p1:=plot(f(t),t=-1..3,thickness=3,discont=true,color=black):
> p2:= line([1,-1], [1,1], color=grey, linestyle=3):
> p3:= line([2,-1], [2,0], color=grey, linestyle=3):
> plots[display](p1,p2,p3);
```



Дальнейшие действия аналогичны изложенным в предыдущих примерах.

```
> laplace(odu,t,p);
```

$$p^2\mathrm{laplace}(y(t),t,p)+\mathrm{laplace}(y(t),t,p)=\frac{1}{p}-\frac{2e^{-p}}{p}+\frac{e^{-2p}}{p}$$

```
> subs(laplace(y(t),t,p)=Y(p),%);
```

$$p^2Y(p)+Y(p)=\frac{1}{p}-\frac{2e^{-p}}{p}+\frac{e^{-2p}}{p}$$

```
> Y(p):=solve(%,Y(p));
```

$$Y(p):=\frac{-1+2e^{-p}-e^{-2p}}{p(p^2+1)}$$

```
> y:=t->invlaplace(Y(p),p,t): y=y(t);
```

$$y=1-\cos(t)-4\,\mathrm{Heaviside}\,(t-1)\sin\left(\frac{t}{2}-\frac{1}{2}\right)^2+2\,\mathrm{Heaviside}(t-2)\sin\left(\frac{t}{2}-1\right)^2$$

Аналогично решаются и системы дифференциальных уравнений.

Пример 25. *Решить систему уравнений*

$$\begin{cases} x'=y+z, & x(0)=3,\\ y'=x+z, & y(0)=-1,\\ z'=x+y, & z(0)=2. \end{cases}$$

Зададим условия задачи:

```
> sys:=diff(x(t),t)=y(t)+z(t),
> diff(y(t),t)=x(t)+z(t),
> diff(z(t),t)=x(t)+y(t);
> ic:=x(0)=3, y(0)=-1, z(0)=2;
```

$$sys := \frac{d}{dt}\,x(t) = y(t) + z(t),\ \frac{d}{dt}\,y(t) = x(t) + z(t),\ \frac{d}{dt}\,z(t) = x(t) + y(t)$$

$$ic := x(0) = 3,\ y(0) = -1,\ z(0) = 2$$

1) Решим систему без указания метода решения.

```
> dsolve(sys,ic,x(t),y(t),z(t));
```

$$\{z(t) = \frac{4}{3}e^{2t} + \frac{2}{3}e^{-t},\ y(t) = \frac{4}{3}e^{2t} + \frac{7}{3}e^{-t},\ x(t) = \frac{4}{3}e^{2t} + \frac{5}{3}e^{-t},\}$$

2) Решим систему операционным методом.

```
> dsolve(sys,ic,x(t),y(t),z(t),laplace);
```

$$\{z(t) = \frac{4}{3}e^{2t} + \frac{2}{3}e^{-t},\ y(t) = \frac{4}{3}e^{2t} + \frac{7}{3}e^{-t},\ x(t) = \frac{4}{3}e^{2t} + \frac{5}{3}e^{-t},\}$$

3) Посмотрим теперь подробное решение этой системы операционным методом.

```
> with(inttrans):
> sys:=diff(x(t),t)=y(t)+z(t),
> diff(y(t),t)=x(t)+z(t),
> diff(z(t),t)=x(t)+y(t);
> x(0):=3, y(0):=-1, z(0):=2;
```

$$sys := \frac{d}{dt}\,x(t) = y(t) + z(t),\ \frac{d}{dt}\,y(t) = x(t) + z(t),\ \frac{d}{dt}\,z(t) = x(t) + y(t)$$

$$x(0) := 3,\ y(0) := -1,\ z(0) := 2$$

```
> laplace(sys,t,p);
```

$$\begin{gathered}\{p\ \mathrm{laplace}(x(t),t,p) - 3 = \mathrm{laplace}(y(t),t,p) + \mathrm{laplace}(z(t),t,p),\\ p\mathrm{laplace}(z(t),t,p) - 2 = \mathrm{laplace}(x(t),t,p) + \mathrm{laplace}(y(t),t,p),\\ p\ \mathrm{laplace}(y(t),t,p) + 1 = \mathrm{laplace}(x(t),t,p) + \mathrm{laplace}(z(t),t,p)\}\end{gathered}$$

```
> subs({laplace(x(t),t,p)=X(p),laplace(y(t),t,p)=Y(p),
  laplace(z(t),t,p)=Z(p)},%);
```

$$\{pX(p) - 3 = Y(p) + Z(p), pZ(p) - 2 = X(p) + Y(p), pY(p) + 1 = X(p) + Z(p)\}$$

```
> solve(%,X(p),Y(p),Z(p)); assign(%);
```

$$\{Z(p) = \frac{2p}{-p-2+p^2},\ X(p) = \frac{3p-2}{-p-2+p^2},\ Y(p) = -\frac{-6+p}{-p-2+p^2}\}$$

```
> X(p):=convert(X(p),parfrac,p);
> Y(p):=convert(Y(p),parfrac,p);
> Z(p):=convert(Z(p),parfrac,p);
```

$$X(p) := \frac{5}{3(p+1)} + \frac{4}{3(p-2)}$$
$$Y(p) := -\frac{7}{3(p+1)} + \frac{4}{3(p-2)}$$
$$Z(p) := \frac{2}{3(p+1)} + \frac{4}{3(p-2)}$$

Получили ответ:

```
> x:=t->invlaplace(X(p),p,t): x=x(t);
> y:=t->invlaplace(Y(p),p,t): y=y(t);
> z:=t->invlaplace(Z(p),p,t): z=z(t);
```

$$x = \frac{5}{3}e^{(-t)} + \frac{4}{3}e^{(2t)}$$
$$y = -\frac{7}{3}e^{(-t)} + \frac{4}{3}e^{(2t)}$$
$$z = \frac{2}{3}e^{(-t)} + \frac{4}{3}e^{(2t)}$$

Убедимся в том, что он удовлетворяет начальным условиям:

```
> x0=x(0); y0=y(0); z0:=z(0);
```

$$x0 = 3$$
$$y0 = -1$$
$$z0 := 2$$

7.2 Задачи для самостоятельного решения

1. Используя методы операционного исчисления, решить задачу Коши для обыкновенных дифференциальных уравнений.

1)	$y'' + 4y' + 29y = e^{-2x}$,	$y(0) = 0$,	$y'(0) = 1$
2)	$y'' + 2y' = \sin\frac{x}{2}$,	$y(0) = -2$,	$y'(0) = 4$
3)	$y'' - 2y' - 3y = 2x$,	$y(0) = 1$,	$y'(0) = 1$
4)	$y'' + 4y = \sin 2x$,	$y(0) = 1$,	$y'(0) = 1$
5)	$y'' - 3y' + 2y = e^x$,	$y(0) = 1$,	$y'(0) = 0$
6)	$y'' + y' + y = x^2 + x$,	$y(0) = 1$,	$y'(0) = -3$
7)	$2y'' + 3y' + y = 3e^x$,	$y(0) = 0$,	$y'(0) = 1$
8)	$2y'' + 5y' = 29\cos x$,	$y(0) = -1$,	$y'(0) = 0$
9)	$y'' + 4y = 8\sin 2x$,	$y(0) = 3$,	$y'(0) = -1$
10)	$y'' - y' - 6y = 2$,	$y(0) = 1$,	$y'(0) = 0$
11)	$y'' + 4y = 4e^{2x} + 4x^2$,	$y(0) = 1$,	$y'(0) = 2$
12)	$y'' + 4y' + 4y = x^3e^{2x}$,	$y(0) = 1$,	$y'(0) = 2$
13)	$y'' + 2y' = \sin\frac{x}{2}$,	$y(0) = -2$,	$y'(0) = 4$
14)	$y'' - 2y' - 3y = 2x$,	$y(0) = 1$,	$y'(0) = 1$
15)	$y'' + 4y = \sin 2x$,	$y(0) = 1$,	$y'(0) = 1$
16)	$y'' + y = \cos x + \sin 2x$,	$y(0) = 0$,	$y'(0) = 0$

2. Решить систему уравнений.

1) $\begin{cases} x' = x + 3y + 2 & x(0) = -1, \\ y' = x - y + 1 & y(0) = 2. \end{cases}$
2) $\begin{cases} x' = x + y + e^t & x(0) = 0, \\ y' = 3x - y & y(0) = 0. \end{cases}$

3) $\begin{cases} x' = y + 1 & x(0) = 0, \\ y' = 3x + 2y & y(0) = 0. \end{cases}$
4) $\begin{cases} x' = -x + 3y + 1 & x(0) = 1, \\ y' = x + y & y(0) = 2. \end{cases}$

5) $\begin{cases} x' = x + 4y & x(0) = 1, \\ y' = 2x - y + 9 & y(0) = 0. \end{cases}$
6) $\begin{cases} x' = -x + 2y + 1 & x(0) = 0, \\ y' = 4x - y & y(0) = 1. \end{cases}$

7) $\begin{cases} x' = 2x + 5y & x(0) = 1, \\ y' = x - 2y + 2 & y(0) = 1. \end{cases}$
8) $\begin{cases} x' = -2x + 5y + 1 & x(0) = 0, \\ y' = x + y + 1. & y(0) = 2. \end{cases}$

9) $\begin{cases} x' = 3x + 4y & x(0) = 2, \\ y' = -5x - 3y + 2 & y(0) = 0. \end{cases}$
10) $\begin{cases} x' = -3x - 4y + 1 & x(0) = 0, \\ y' = 2x + 3y & y(0) = 2. \end{cases}$

11) $\begin{cases} x' = -2x + 6y + 1 & x(0) = 0, \\ y' = 2x + 2 & y(0) = 1. \end{cases}$
12) $\begin{cases} x' = 2x + 3y + 1 & x(0) = -1, \\ y' = 4x - 2y & y(0) = 0. \end{cases}$

13) $\begin{cases} x' = x + 2y & x(0) = 0, \\ y' = 2x + y + 1 & y(0) = 5. \end{cases}$
14) $\begin{cases} x' = 2x - 2y & x(0) = 3, \\ y' = -4x & y(0) = 1. \end{cases}$

15) $\begin{cases} x' = -x - 2y + 1 & x(0) = 1, \\ y' = -3/2\, x + y & y(0) = 0. \end{cases}$
16) $\begin{cases} x' = 3x + 5y + 2 & x(0) = 0, \\ y' = 3x + 3y + 1 & y(0) = 2. \end{cases}$

3. Решить задачу Коши, если функция $f(t)$ задана графически:

1) $$y''(t) + 2y'(t) + y(t) = f(t),\quad y(0) = 0,\ y'(0) = 0,$$



2) $$y''(t) + y(t) = f(t),\quad y(0) = 0,\ y'(0) = 0,$$



3) $$y''(t) - y(t) = f(t),\quad y(0) = 0,\ y'(0) = 0,$$



4) $$y''(t) + 4y(t) = f(t),\quad y(0) = 0,\ y'(0) = 0,$$



5) $$y''(t) + 2y'(t) + y(t) = f(t),\quad y(0) = 0,\ y'(0) = 0,$$



6) $y''(t) + y(t) = f(t),$
$y(0) = 0,\ y'(0) = 0,$



7) $y''(t) - y(t) = f(t),$
$y(0) = 0,\ y'(0) = 0,$



8) $y''(t) + 4y(t) = f(t),$
$y(0) = 0,\ y'(0) = 0,$



9) $y''(t) + 2y'(t) + y(t) = f(t),$
$y(0) = 0,\ y'(0) = 0,$



10) $y''(t) + y(t) = f(t),$
$y(0) = 0,\ y'(0) = 0,$



11) $y''(t) - y(t) = f(t),$
$y(0) = 0,\ y'(0) = 0,$



12) $y''(t) + 4y(t) = f(t),$
$y(0) = 0,\ y'(0) = 0,$



13) $y''(t) + 2y'(t) + y(t) = f(t),$
$y(0) = 0,\ y'(0) = 0,$



14) $y''(t) + y(t) = f(t),$
$y(0) = 0,\ y'(0) = 0,$



15) $y''(t) - y(t) = f(t),$
$y(0) = 0,\ y'(0) = 0,$



16) $y''(t) + 4y(t) = f(t),$
$y(0) = 0,\ y'(0) = 0,$



ЛИТЕРАТУРА.

1. *Дьяконов В.П.* Maple 9 в математике, физике и образовании. – М.:СОЛОН-Пресс, 2004. – 688 с.
2. *Матросов А.В.* Maple 6 Решение задач высшей математики и механики. – БХВ-Санкт-Петербург, 2001. – 528 с.
3. *Сдвижков О.А.* Математика на компьютере: Maple 8, –М.: СОЛОН-Пресс, 2003. – 176 с.
4. *Манзон Б.М.* Maple V power edition, – М: Информационно-издательский дом Филинъ, 1998. – 240 с.
5. *Ильин В. А., Позняк Э. Г.* Аналитическая геометрия, – М.: ФИЗМАТЛИТ, 2004. -— 224 с.
6. *Лаврентьев М.А. Шабат Б.В.* Методы теории функций комплексной переменной, – М.: Наука, 1987. — 688 с.
7. *Сидоров Ю.В., Федорюк М.В., Шабунин М.Н.* Лекции по теории функций комплексной переменной, –М.: Наука, 1989. – 477 с.
8. *Соломенцев Е.Д.* Функции комплексного переменного и их применения, – М.: Высшая школа, 1988. – 167 с.
9. *Гольдберг А.А., Шеремета М.М., Заблоцький М.В., Скасків О.Б.* Комплексний аналіз. – Львів: Афіша, 2002. – 204 с.
10. *Свешников А.Г., Тихонов А.Н.* Теория функций комплексной переменной, – М.: Наука, 1974. – 319 с.
11. *Парфёнова Н.Д.* Комплексный анализ, – ХНУ имени В.Н. Каразина, 2008. – 80 с.

Навчальне видання

Клочко Тамара Володимирівна
Парфьонова Наталія Дмитрівна

РІШЕННЯ ЗАДАЧ КОМПЛЕКСНОГО АНАЛІЗУ ЗАСОБАМИ Maple

Навчально-методичний посібник
з математичного моделювання для студентів 2-го курсу
фізичного та радіо-фізичного факультетів

Друкується в авторській редакції
Макет обкладинки І.М.Дончик

61077, Харків майдан Свободи, 4, Харківський національний університет
імені В.Н. Каразіна, організаційно-видавничий відділ НМЦ

Підписано до друку 28.03.08. Формат 60×84/16.
Папір офсетний. Друк ризографічний.
Обл.-вид. арк. 4,3. Умов.-друк. арк. 3,7.
Наклад 50 прим. Ціна договірна.